
/*=================================================================== 

Covance Study ID  : COV-000000106331 

Program Name        : d_adDE.sas  

Purpose                : Program to ADDE dataset 

Author              : siva karnati 

Date of Creation        : 23MAR2015   

Input Data    : SDTM.DE,SDTM.SUPPDE,ADAM.ADSL   

Output Data    : ADAM.ADDE 

Macros Called   : m_printto,m_attrib_adam,m_logchk 

===================================================================== 

Modification History    

===================================================================== 

Modified by           :    

Modification Date     : 

Modification Descrption  :  

 

===================================================================*/ 

%m_printto(route=YES); 

 

libname adam "&base2/datasets/adam/cleaned_adam";  

 

 

/* SUPPDE */ 

 

PROC SORT DATA=SDTM.SUPPDE OUT=SUPPDE;BY USUBJID IDVARVAL;RUN; 



 

 

PROC TRANSPOSE DATA=SUPPDE OUT=SUPPDE_(DROP=_:); 

 BY USUBJID IDVARVAL; 

 VAR QVAL; 

 ID QNAM; 

RUN; 

/*DE DATA*/ 

 

DATA DE; 

 SET SDTM.DE; 

RUN; 

 

PROC SQL; 

 CREATE TABLE DE_ AS SELECT A.*,AEREL,NDSN FROM DE A, SUPPDE_ B WHERE 

A.USUBJID=B.USUBJID  AND A.DESEQ=INPUT(B.IDVARVAL,2.); 

QUIT; 

 

DATA DE_1; 

 SET DE_(RENAME=(AEREL=AEREL_ NDSN=NDSN_ )); 

 FORMAT  ASTDT AENDT DATE9. ASTDTM AENDTM DATETIME13. AEREL NDSN AENUM $11. 

AVISIT $40.  ; 

 LENGTH DESEVN AVISITN 8.; 

  IF STRIP(DESEV)="MAJOR" THEN DESEVN=2 ; 

   ELSE DESEVN=1; 

     AEREL=STRIP(AEREL_); 



  AENUM=""; 

  NDSN=STRIP(NDSN_); 

  AVISIT=STRIP(VISIT); 

  AVISITN=VISITNUM; 

  IF LENGTH(DESTDTC)>10 THEN ASTDTM=INPUT((DESTDTC||":00"),IS8601DT.); 

   ELSE IF LENGTH(DESTDTC)=10 THEN 

ASTDTM=DHMS(INPUT(DESTDTC,YYMMDD10.),0,0,0); 

  IF ASTDTM NE . THEN  ASTDT=DATEPART(ASTDTM); 

  IF LENGTH(DEENDTC)>10 THEN AENDTM=INPUT((DEENDTC||":00"),IS8601DT.); 

   ELSE  AENDTM=DHMS(INPUT(DEENDTC,YYMMDD10.),0,0,0); 

   IF AENDTM NE . THEN  AENDT=DATEPART(AENDTM); 

 

 

 KEEP USUBJID DESEQ DESPID SPDEVID DETERM DEDECOD DECAT DESEV DESEVN DEACNDEV 

AEREL NDSN  

   AENUM AVISIT AVISITN DESTDTC ASTDT ASTDTM  AENDTM DESTDY DEENDTC AENDT 

DEENDY EPOCH ;  

 

RUN; 

 

/*ADSL DATA*/ 

 

DATA ADSL; 

 SET ADAM.ADSL(WHERE=(DTESTDT NE .)); 

 KEEP STUDYID USUBJID SUBJID SUBJIDN SITEID AGE SEX SEXC SEXN RACE DTHFL HEIGHT 

WEIGHTBL 

      BMI UCPDGR1  UCPDGR1N NICOGR1 NICOGR1N  TARGR1  TARGR1N ENRLFL SCRFFL EXFL 



   EXNOTRFL ENFL COMPLFL FUPFL /*SAFFL FSAFFL*/ FSAFBFL FSAFAFL SAFBFL SAFAFL  

FASFL PPROT1FL PPROT2FL PPROT3FL 

  PPROT4FL  RANDFL TRTSDTM TRTSTMF TRTSDT TRTSDAY TRTEDTM TRTETMF TRTEDT 

TRTEDAY 

  TRT01P TRT01PN TRT01A TRT01AN RANDDT 

 ; 

RUN; 

/*COMBINE ADSL  DE DATA*/ 

PROC SORT DATA=ADSL; BY USUBJID ; RUN; 

PROC SORT DATA=DE_1;BY USUBJID;RUN; 

 

DATA DE_2; 

 MERGE ADSL(IN=A) DE_1(IN=B); 

  BY USUBJID; 

 FORMAT ANYDEFL  $2. APERIODC $10. TRTP TRTA  $40.; 

 LENGTH  APERIOD ASTDAY AENDAY  TRTPN  TRTAN 8.; 

  IF B THEN ANYDEFL="Y"; 

   ELSE ANYDEFL="N"; 

  IF NMISS(TRTSDT,ASTDT)=0 THEN DO; 

   ASTDAY=ASTDT-TRTSDT+1; 

  END; 

  IF NMISS(TRTSDT,AENDT)=0 THEN DO; 

   AENDAY=AENDT-TRTSDT+1; 

  END; 

  APERIOD=1; 

  APERIODC="Period 1"; 



  TRTP=STRIP(TRT01P); 

  TRTPN=TRT01PN; 

  TRTA=STRIP(TRT01A); 

  TRTAN=TRT01AN; 

 

DROP TRT01:;  

   

RUN; 

 

/* GET SV DATES FOR VISITS 'DAY 6/DISCHARGE CONFINEMENT DAY 91/DISCHARGE AMBULATORY*/ 

 

DATA SV; 

 SET SDTM.SV;  

 FORMAT VISIT_ $10.SVSTDTC_ DATE9. ; 

  WHERE VISIT IN ("DAY 6/DISCHARGE CONFINEMENT","DAY 91/DISCHARGE 

AMBULATORY"); 

  VISIT_=STRIP(SCAN(VISIT,1,"/")); 

  IF SVSTDTC NE "" THEN SVSTDTC_=INPUT(SVSTDTC,YYMMDD10.); 

 KEEP USUBJID VISIT SVSTDTC VISIT_ SVSTDTC_; 

RUN; 

 

PROC SORT DATA=SV ; BY USUBJID; RUN; 

 

PROC TRANSPOSE DATA=SV OUT=SV1(DROP=_:); 

 BY USUBJID; 

 VAR SVSTDTC_; 



 ID VISIT_; 

RUN; 

 

PROC FORMAT ; 

 

VALUE $ASPR 

 "Pre-Randomization Period" =1 

 "Confinement Period"=2 

 "Ambulatory Period"=3 

 "Safety Follow-up Period"=4 

; 

RUN; 

 

DATA ADDE; 

 MERGE DE_2(IN=A) SV1(IN=B); 

 BY USUBJID;  

 FORMAT ASPERC $40.; 

 LENGTH ASPER 8.; 

 IF A; 

 

 if trtpn in (97 98) then do;  

 asperc= 'Pre-Randomization Period';  

 end;  

  

  IF .<AVISITN <101 THEN ASPERC="Pre-Randomization Period"; 



   ELSE IF  101<=AVISITN<=106 THEN ASPERC="Confinement Period"; 

  ELSE IF 106<AVISITN<=191 THEN ASPERC="Ambulatory Period"; 

    ELSE IF AVISITN>191 THEN ASPERC="Safety Follow-up Period"; 

    ASPER=INPUT(PUT(ASPERC,$ASPR.),BEST.); 

   DROP DAY_6 DAY_91 RANDDT; 

   FORMAT AEREL AENUM NDSN ASPER ASPERC AVISIT APERIODC TRTP TRTA ; 

RUN; 

 

%m_attrib_adam(dset=ADDE); 

 

PROC SORT DATA=ADDE OUT=ADAM.ADDE(LABEL="Device Events Analysis Dataset"); 

BY USUBJID DETERM DEDECOD DECAT DESTDTC DESPID; 

RUN; 

 

proc compare base=adam.adde compare=qadam.qadde listall; 

run; 

%m_logchk; 

 

 

  

 

 

 

 

 



 

 

 

 

  

 

 

 

 


