
 

/*===================================================================================

==============* 

| Covance Study Number    : 000000106331                                                           | 

| Program Name            : adds.sas                                                               | 

| Purpose                 : Create Adam Dataset (ADDS)                                             | 

| Input Data              : adsl ,sdtm.DS                                                               | 

| Output Data             : adma.adds                                                              |         | 

|                                                                                                  | 

| Macros Called           :%m_printto , %m_logchk, %m_attrib_adam                                                                     

| 

| Originally Performed by :paddepalli                                                              | 

| Date                    : 13Mar2015                                                              | 

|====================================================================================

==============| 

| Modification History : Original Version                                                          | 

|--------------------------------------------------------------------------------------------------| 

| Modified by          :                                                                           | 

| Modification Date    :                                                                           | 

| Modification Reason  :                                                                           | 

+====================================================================================

==============*/ 

* macro to save output and log to appropriate areas ; 

 

 

libname adam "&base2/datasets/adam/cleaned_adam";  

 



libname sdtm "/cvn/projects/prj/data/000000106331/datasets/sdtm/sdtmx"; 

%m_printto(route=YES); 

options  validvarname=upcase missing=' '; 

 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

****************; 

* bring in ADSL ; 

****************; 

 

data adsl; 

 set adam.adsl; 

run; 

 

************************; 

* bring in SUPPDS ; 

************************; 

proc transpose data = sdtm.suppds out=suppds(drop = _:); 

 by usubjid idvarval; 

 var qval; 

 id qnam; 

 idlabel qlabel; 

run; 

 



data suppds2(drop = idvarval); 

 set suppds; 

 dsseq = input(idvarval,best.); 

run; 

 

proc sort data=suppds2; 

 by usubjid dsseq; 

run; 

 

****************; 

* bring in DS   ; 

****************; 

proc sort data = sdtm.ds out = ds1; 

 by usubjid dsseq; 

run; 

 

data ds2; 

length other $200; 

 merge ds1 suppds2 (rename=(other=other_)); 

 by usubjid dsseq; 

 format astdtm datetime13. adt astdt date9. asttm time5.  ; 

 

 if length(dsstdtc) gt 10 then astdtm = input(dsstdtc,e8601dt.); 

 if not missing (dsdtc) then  

 adt = input(dsdtc,yymmdd10.); 



   other=other_; 

 if not missing(astdtm) then astdt = datepart(astdtm); 

 else if length(dsstdtc) = 10 then astdt = input(dsstdtc,yymmdd10.); 

 

 if not missing(astdtm) then asttm = timepart(astdtm); 

 

 keep usubjid dsseq dsterm dsdecod dscat dsdtc dsstdtc dsstdy adt astdtm astdt asttm epoch 

other;  

run; 

 

***************************; 

* Combine ADSL and DS data *; 

***************************; 

proc sort data= adsl;by usubjid;run; 

proc sort data=ds2;by usubjid;run; 

data slds(drop=trtsdt);   

 merge adsl ds2(in = a); 

 by usubjid; 

 if a;  * only include subjects with DS data ; 

 format aday astday aperiod trtan trtpn 8. trta trtp $40. aperiodc $10.;   

 if not missing(adt) and not missing(trtsdt)  then do; 

 aday = adt - trtsdt + 1; 

 end; 

 if  not missing(astdt) and  not missing(trtsdt)   then  

 astday = astdt - trtsdt + 1; 

 



 aperiod=1; 

 if not missing(aperiod) then do; 

  aperiodc = 'Period ' || put(aperiod,1.); 

 end; 

  if aperiod=1 then do; 

 TRTP= TRT01p; 

 TRTPN=trt01pn; 

  TRTA=trt01a; 

  trtan=trt01an; 

 end; 

run; 

************************; 

* create output dataset ; 

************************; 

data adds; 

 set slds; 

 run; 

 

 

%m_attrib_adam(dset=ADDS); 

 

proc sort data = adds out = adam.adds(label= 'Disposition Analysis Dataset'); 

 by usubjid dsstdtc dsdtc dsdecod dsterm; 

run; 

 



*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

data pr; 

 set adds; 

 run; 

data qc; 

 set qadam.qadds; 

 run; 

proc compare base = adam.adds compare = qadam.qadds listobs listvar; 

run; 

 

%m_logchk; 

 


