
/*=================================================================================* 

| Covance Study Number    : COV-  106343                                           | 

| Program Name            : d_2adex.sas                                            | 

| Purpose                 : create ADEX                                            | 

| Input Data              : ADAM.ADSL,SDTM.EX,SDTM.DX,SDTM.SU           | 

| Output Data             : ADAM.ADEX                                              | 

|                                                                                  | 

| Macros Called           :                                                        | 

|                                                                                  | 

| Originally Performed by : Serona Zheng                                           | 

| Date/Time billed        : 10Apr2015                                              | 

|                                                                                  | 

+==================================================================================+ 

|                                                                                  | 

| Modification History    :                                                        | 

|                                                                                  | 

| Programmer              : Serona Zheng                                           | 

| Date                    : 21May2015                                              | 

| Reason for Change       : Change -dose to -dostxt for -9 or -1 values            | 

|                                                                                  | 

| Programmer              : Serona Zheng                                           | 

| Date                    : 05Jun2015                                              | 

| Reason for Change       : Change svstdtc to svendtc based on John's email on 05Jun2015| 

|                                                                                  | 

| Programmer              : Serona Zheng                                           | 



| Date                    : 30Jun2015                                              | 

| Reason for Change       : Fix the typo in param 'Daily Electronic Cigarette'     | 

+=================================================================================*/ 

 

libname adam "&base2/datasets/adam/cleaned_adam";  

 

libname sdtm "/cvn/projects/prj/data/000000106343/datasets/sdtm/sdtmx";  

/*libname sdtm "/cvn/projects/prj/data/000000106343/datasets/sdtm"; */ 

 

 

%m_printto(route=YES); 

 

***Create macro to assign apuper; 

%macro apuper(dsin1=,dsout1=); 

data &dsout1; 

 merge &dsin1(in=a) ds sv1; 

 by usubjid; 

 if a; 

 if . < astdt <=randdt then apuper = 0; 

 else if nmiss(randdt,astdt) = 0 and randdt < astdt and index(parcat2,"PRODUCT USE DIARY")=0 
and avisitn < 130 then apuper = 1; 

 else if index(parcat2,"PRODUCT USE DIARY") and ((v30 ne . and . < astdt <= v30) or (v30 = . and 
/*6<=*/astday <=31)) then apuper = 2; 

 else if index(parcat2,"PRODUCT USE DIARY") and ((v60 ne . and v30 < astdt <= v60) or (v30 ne . 
and v60 = . and 31 < astday <=61) or (v30 eq . and v60 = . and 31 < astday <=61)) then apuper = 3; 

 else if index(parcat2,"PRODUCT USE DIARY") and ((v60 > . and astdt > v60) or (v60 eq . and 
astday > 61))  then apuper = 4; 



 if nmiss(astdt,v91) eq 0 and astdt > v91 then put "User warning: " usubjid " take drug date is 
greate than study end date"; 

run; 

%mend; 

 

%macro cal_1(dsin2=,dsout2=,apuper=); 

 

proc sql; 

 create table temp as 

  %if &apuper = 1 %then %do; 

   select usubjid,paramn,apuper,parcat1, sum(aval) as aval1 

   from &dsin2 

   where apuper >=2 

   group by usubjid,paramn,apuper,parcat1; 

  %end; 

  %else  %do; 

   select usubjid,paramn,parcat1,sum(aval) as aval1 

   from &dsin2 

   where apuper >=2 

   group by usubjid,paramn,parcat1; 

  %end; 

quit; 

 

data &dsout2; 

 merge temp(in=a) sv_d(keep=usubjid d_2 d_3 d_4 d_dis); 

 by usubjid; 



 if a; 

 %if &apuper = 1 %then %do; 

  if apuper = 2 and nmiss(aval1,d_2) = 0 and d_2 > 0 then aval = aval1/d_2; 

  else if apuper = 3 and nmiss(aval1,d_3) = 0 and d_3 > 0 then aval = aval1/d_3; 

  else if apuper = 4 and nmiss(aval1,d_4) = 0 and d_4 > 0 then aval = aval1/d_4; 

 %end; 

 %else %do; 

 if nmiss(aval1,d_dis) = 0 and d_dis > 0 then aval = aval1/d_dis; 

 %end; 

run; 

%mend; 

 

***Get data from adsl; 

proc sort data=adam.adsl out=adsl;by usubjid;run; 

 

***Get randdt from ds; 

data ds; 

 set adsl(keep=usubjid randdt where=(randdt ne .)); 

run; 

 

***Get visit date from SV; 

data sv; 

 set sdtm.sv; 

 svdt = input(svendtc, is8601da.); 

 format svdt date9.; 



run; 

 

proc transpose data=sv(where=(visitnum in (130 /*131 161*/ 160 190 191 106)))  

    out=sv1(drop=_name_ rename=(v130=v30 /*v131=v31 v161=v61*/ 
v160=v60 v190=v90 v191=v91 v106=v6)) prefix=v; 

 by usubjid; 

 var svdt; 

 id visitnum; 

run; 

 

***Get last visit; 

data sv_l1; 

 set adsl; 

 lastdt =lvisdt; 

 keep usubjid lastdt; 

 format lastdt date9.; 

run; 

 

data sv_d; 

 merge sv1(in=a) sv_l1; 

 by usubjid; 

 if a; 

 if v30 ne . then d_2=v30-v6+1; 

 if v30 = . and v91 ne . then d_2 = v91-v6; 

 if v30 = . and v91 eq . then d_2 = lastdt-v6; 

 



 if v60 = . and v30 ne . then do; 

  if v91 ne . then d_3 = v91-1-v30; 

  else if v91 eq . then d_3 = lastdt-1-v30; 

 end; 

 if v60 ne . and v30 = . then do; 

  d_3 = v60-(v6+24); 

 end; 

 

 if v60 ne . and v30 ne . then d_3 = v60-v30; 

 

 if nmiss(V91,v60) = 0 then d_4 = v91-v60-1; 

 else if v91 = . and v60 ne . then d_4 = lastdt-v60 -1; 

 else if v91 ne . and v60 = . then d_4 = v91-1-(v6+54); 

 else if v91 = . and v60 = . then d_4 = lastdt-1-(v6+54); 

 

 if nmiss(v6,v91) eq 0 then d_dis = v91-v6; 

 if v6 ne . and v91 = . then d_dis=lastdt-v6; 

run; 

 

*******************************************************************************; 

 

************Step1: Get data and create parameter from DX                *******;     

 

*******************************************************************************; 

 



***Get data from DX; 

data dx; 

 length srcdom $2 desc reasnd $200 parcat1 parcat2 $100 avalu status $20 lot $7 astdtc aendtc 
$16 epoch $25 avisit $40; 

 set sdtm.dx(rename=(epoch=epoch1)); 

 if dxcat = "TOBACCO HEATING SYSTEM MENTHOL";  

 srcdom = 'DX'; 

 srcseq = dxseq; 

 desc = dxtrt; 

 parcat1 = dxcat; 

 parcat2 = dxscat; 

 if dxdostxt = '-9' then aval = 0; 

 else if dxdostxt = '-1' then aval = .; 

 else aval = dxdose; 

 avalu = dxdosu; 

 status = dxstat; 

 reasnd = dxreasnd; 

 lot = dxlot; 

 avisitn = visitnum; 

 avisit = visit; 

 astdtc = dxstdtc; 

 aendtc = dxendtc; 

 astdy = dxstdy; 

 aendy = dxendy; 

 epoch = epoch1; 

 



 

  if ASTDTC^='' then astdt = input(scan(ASTDTC,1,"T"),yymmdd10.); 

  if index(ASTDTC,"T") then do; 

      if length(scan(ASTDTC,2,"T")) = 5 then do; 

        astdtm = dhms(astdt,0,0,input(scan(ASTDTC,2,"T"),time5.)+0); 

      end; 

      else if length(scan(ASTDTC,2,"T")) = 8 then astdtm = 
dhms(astdt,0,0,input(scan(ASTDTC,2,"T"),time8.)); 

      else put "WAR" "NING: Unexpected time format " ASTDTC=; 

    end; 

/*  else astdtm = dhms(astdt,0, 0, 0);*/ 

  if AENDTC^='' then aendt = input(scan(AENDTC,1,"T"),yymmdd10.); 

    if index(AENDTC,"T") then do; 

      if length(scan(AENDTC,2,"T")) = 5 then do; 

        aendtm = dhms(aendt,0,0,input(scan(AENDTC,2,"T"),time5.)+0); 

      end; 

      else if length(scan(AENDTC,2,"T")) = 8 then aendtm = 
dhms(aendt,0,0,input(scan(AENDTC,2,"T"),time8.)); 

      ELSE IF LENGTH(SCAN(AENDTC,2,"T")) = 2 THEN AENDTM = .; 

      else put "WAR" "NING: Unexpected time format " AENDTC=; 

    end; 

 

 format astdtm aendtm datetime13. astdt aendt date9.; 

run; 

 

***Calculate astday; 



proc sort data=dx;by usubjid;run; 

data dx; 

 merge dx(in=a) adsl(keep=usubjid trtsdt); 

 by usubjid; 

 if a; 

 if nmiss(astdt,trtsdt) = 0 then astday = astdt-trtsdt + 1; 

run; 

 

***Assign apuper; 

%apuper(dsin1=dx,dsout1=dx) 

 

***Derive param,parmcd,paramn from dx; 

***create parameter 0; 

data dx_p0; 

 length param $100 paramcd $8; 

 set dx; 

 if index(dxscat,"PRODUCT USE DIARY")=0; 

  paramn = 0; 

  paramcd = "THS2_2"; 

  param = "THS 2.2"; 

run; 

***create parameter 1; 

data dx_1; 

 set dx; 

 if index(dxscat,"PRODUCT USE DIARY")=0 and avisitn not in (130 160 190); 



 keep usubjid astdtc astdt aendtc aendt astdy aendy dxdose apuper avisitn avisit parcat1 parcat2 
astday epoch; 

run; 

 

proc sql; 

 create table dx_1_ as 

  select usubjid,astdt, sum(dxdose) as aval  

   from dx_1 

   group by usubjid,astdt; 

quit;  

 

proc sort data=dx_1(drop=dxdose) nodupkey;by usubjid astdt astdtc aendtc aendt astday aendy apuper 
epoch;run;  

 

data dx_1; 

 set dx_1; 

 by usubjid astdt; 

 if last.astdt; 

 astdtc = scan(astdtc,1,'T'); 

 aendtc = scan(aendtc,1,'T'); 

run; 

 

data dx_p1; 

 length dtype $20; 

 merge dx_1(in=a) dx_1_; 

 by usubjid astdt; 



 if a; 

 dtype = "SUM"; 

run; 

 

data dx_p1; 

 length param $100 paramcd $8 parcat3 $100 dtype avalu $20; 

 set dx(where=(index(dxscat,"PRODUCT USE DIARY"))) dx_p1; 

 param = "Daily THS 2.2 Administration"; 

 paramcd = "DTHS2_2"; 

 paramn = 1; 

 avalu = "STICKS/DAY"; 

 parcat3 = "DAILY PRODUCT USE"; 

 parcat3n = 1; 

run; 

 

***create parameter 2; 

%cal_1(dsin2=dx_p1,dsout2=dx_p2,apuper=1) 

 

 

data dx_p2; 

 length param $100 paramcd $8 parcat3 $100 paramtyp $10 avalu $20; 

 set dx_p2; 

 param = "Average Daily THS 2.2 by Period"; 

 paramcd = "PDTHS2_2"; 

 paramn = 2;  



 parcat3 = "AVERAGE DAILY PRODUCT USE BY PERIOD"; 

 parcat3n = 2; 

 paramtyp = "DERIVED"; 

 avalu = "STICKS/DAY"; 

 dtype = "AVERAGE"; 

 keep usubjid aval apuper param paramcd paramn parcat3 parcat3n paramtyp avalu dtype 
parcat1 ; 

run; 

***create parameter 3; 

%cal_1(dsin2=dx_p1,dsout2=dx_p3,apuper=) 

 

data dx_p3; 

 length param $100 paramcd $8 parcat3 $100 paramtyp $10 avalu dtype $20; 

 set dx_p3; 

 param = "Average Daily THS 2.2 in Ambulatory"; 

 paramcd = "ADTHS2_2"; 

 paramn = 3;  

 parcat3 = "AVERAGE DAILY PRODUCT USE IN AMBULATORY"; 

 parcat3n = 3; 

 paramtyp = "DERIVED"; 

 avalu = "STICKS/DAY"; 

 dtype = "AVERAGE"; 

 keep usubjid aval param paramcd paramn parcat3 parcat3n paramtyp avalu dtype parcat1 ; 

run; 

 

***Combine all parameter together; 



data dx_p; 

 set dx_p0 dx_p1 dx_p2 dx_p3; 

run; 

 

*******************************************************************************; 

 

************Step2: Get data and create parameter from EX                *******;     

 

*******************************************************************************; 

***Get data from EX; 

data ex; 

 length srcdom $2 desc reasnd $200 parcat1 parcat2 $100 avalu status $20 lot $7 astdtc aendtc 
$16 epoch $25 avisit $40; 

 set sdtm.ex(rename=(epoch=epoch1)); 

 if excat = "MENTHOL CONVENTIONAL CIGARETTES";  

 srcdom = 'EX'; 

 srcseq = exseq; 

 desc = extrt; 

 parcat1 = excat; 

 parcat2 = exscat; 

 if exdostxt = '-9' then aval = 0; 

 else if exdostxt = '-1' then aval = .; 

 else aval = exdose; 

 avalu = exdosu; 

 status = exstat; 

 reasnd = exreasnd; 



 lot = ''; 

 avisitn = visitnum; 

 avisit = visit; 

 astdtc = exstdtc; 

 aendtc = exendtc; 

 astdy = exstdy; 

 aendy = exendy; 

 epoch = epoch1; 

 

  if ASTDTC^='' then astdt = input(scan(ASTDTC,1,"T"),yymmdd10.); 

  if index(ASTDTC,"T") then do; 

      if length(scan(ASTDTC,2,"T")) = 5 then do; 

        astdtm = dhms(astdt,0,0,input(scan(ASTDTC,2,"T"),time5.)+0); 

      end; 

      else if length(scan(ASTDTC,2,"T")) = 8 then astdtm = 
dhms(astdt,0,0,input(scan(ASTDTC,2,"T"),time8.)); 

      else put "WAR" "NING: Unexpected time format " ASTDTC=; 

    end; 

  if AENDTC^='' then aendt = input(scan(AENDTC,1,"T"),yymmdd10.); 

    if index(AENDTC,"T") then do; 

      if length(scan(AENDTC,2,"T")) = 5 then do; 

        aendtm = dhms(aendt,0,0,input(scan(AENDTC,2,"T"),time5.)+0); 

      end; 

      else if length(scan(AENDTC,2,"T")) = 8 then aendtm = 
dhms(aendt,0,0,input(scan(AENDTC,2,"T"),time8.)); 

      ELSE IF LENGTH(SCAN(AENDTC,2,"T")) = 2 THEN AENDTM = .; 



      else put "WAR" "NING: Unexpected time format " AENDTC=; 

    end; 

 

 format astdtm aendtm datetime13. astdt aendt date9.; 

run; 

 

***Calculate astday; 

data ex; 

 merge ex(in=a) adsl(keep=usubjid trtsdt); 

 by usubjid; 

 if a; 

 if nmiss(astdt,trtsdt) = 0 then astday = astdt-trtsdt + 1; 

run; 

 

***Assign apuper; 

%apuper(dsin1=ex,dsout1=ex) 

 

***Create parameter from EX; 

***Create parameter 10; 

data ex_p0; 

 length param $100 paramcd $8; 

 set ex; 

 if index(exscat,"PRODUCT USE DIARY")=0; 

  paramn = 10; 

  paramcd = "MCC"; 



  param = "mCC"; 

run; 

 

***Create parameter 11; 

data ex_1; 

 set ex; 

 if index(exscat,"PRODUCT USE DIARY")=0 and avisitn not in (130 160 190); 

 astdtc = put(astdt,yymmdd10.); 

 aendtc = put(aendt,yymmdd10.); 

 keep usubjid astdtc astdt aendtc aendt astdy aendy exdose apuper avisit avisitn parcat1 parcat2 
astday epoch; 

run; 

 

proc sql; 

 create table ex_1_ as 

  select usubjid,astdt, sum(exdose) as aval  

   from ex_1 

   group by usubjid,astdt; 

quit;  

 

proc sort data=ex_1(drop=exdose) nodupkey;by usubjid apuper parcat1 parcat2 astdt astdtc aendt 
aendtc astdy aendy epoch;run;  

data ex_1; 

 set ex_1; 

 by usubjid apuper parcat1 parcat2 astdt astdtc aendt aendtc astdy aendy epoch; 

 if last.astdt; 



run;  

data ex_p1; 

 length dtype $20; 

 merge ex_1(in=a) ex_1_; 

 by usubjid astdt; 

 if a; 

 dtype = "SUM"; 

run; 

data ex_p1; 

 length param $100 paramcd $8 parcat3 $100 avalu $20; 

 set ex(where=(index(exscat,"PRODUCT USE DIARY"))) ex_p1; 

 param = "Daily mCC Administration"; 

 paramcd = "DMCC"; 

 paramn = 11; 

 avalu = "CIGARETTES/DAY"; 

 parcat3 = "DAILY PRODUCT USE"; 

 parcat3n = 1; 

run; 

 

***Create parameter 12; 

%cal_1(dsin2=ex_p1,dsout2=ex_p2,apuper=1) 

 

data ex_p2; 

 length param $100 paramcd $8 parcat3 $100 paramtyp $10 avalu $20; 

 set ex_p2; 



 param = "Average Daily mCC by Period"; 

 paramcd = "PDMCC"; 

 paramn = 12;  

 parcat3 = "AVERAGE DAILY PRODUCT USE BY PERIOD"; 

 parcat3n = 2; 

 paramtyp = "DERIVED"; 

 avalu = "CIGARETTES/DAY"; 

 dtype = "AVERAGE"; 

 keep usubjid aval apuper param paramcd paramn parcat3 parcat3n paramtyp avalu dtype 
parcat1 ; 

run; 

***create parameter 13; 

%cal_1(dsin2=ex_p1,dsout2=ex_p3,apuper=) 

 

data ex_p3; 

 length param $100 paramcd $8 parcat3 $100 paramtyp $10 avalu dtype $20; 

 set ex_p3; 

 param = "Average Daily mCC in Ambulatory"; 

 paramcd = "ADMCC"; 

 paramn = 13;  

 parcat3 = "AVERAGE DAILY PRODUCT USE IN AMBULATORY"; 

 parcat3n = 3; 

 paramtyp = "DERIVED"; 

 avalu = "CIGARETTES/DAY"; 

 dtype = "AVERAGE"; 

 keep usubjid aval param paramcd paramn parcat3 parcat3n paramtyp avalu dtype parcat1 ; 



run; 

 

***Combine all parameter together; 

data ex_p; 

 set ex_p0 ex_p1 ex_p2 ex_p3; 

run; 

 

*******************************************************************************; 

 

************Step3: Get data and create parameter from SU                *******;     

 

*******************************************************************************; 

 

***Get data from SU; 

data su; 

 length srcdom $2 desc reasnd $200 parcat1 parcat2 $100 avalu status $20 lot $7 astdtc aendtc 
$16 epoch $25 avisit $40; 

 set sdtm.su(rename=(epoch=epoch1)); 

 if sucat in ('NRT_USE' 'TOB_USE');  

 srcdom = 'SU'; 

 srcseq = suseq; 

 desc = sutrt; 

 parcat1 = sucat; 

 parcat2 = suscat; 

 if sudostxt = '-9' then aval = 0; 

 else if sudostxt = '-1' then aval = .; 



 else aval = sudose; 

 avalu = sudosu; 

 status = ''; 

 reasnd = ''; 

 lot = ''; 

 avisitn = visitnum; 

 avisit = visit; 

 astdtc = sustdtc; 

 aendtc = ''; 

 astdy = sustdy; 

 aendy = .; 

 epoch = epoch1; 

 

  if ASTDTC^='' then astdt = input(scan(ASTDTC,1,"T"),yymmdd10.); 

  if index(ASTDTC,"T") then do; 

      if length(scan(ASTDTC,2,"T")) = 5 then do; 

        astdtm = dhms(astdt,0,0,input(scan(ASTDTC,2,"T"),time5.)+0); 

      end; 

      else if length(scan(ASTDTC,2,"T")) = 8 then astdtm = 
dhms(astdt,0,0,input(scan(ASTDTC,2,"T"),time8.)); 

      else put "WAR" "NING: Unexpected time format " ASTDTC=; 

    end; 

 

 format astdtm datetime13. astdt date9.; 

run; 

 



***Calculate astday; 

data su; 

 merge su(in=a) adsl(keep=usubjid trtsdt); 

 by usubjid; 

 if a; 

 if nmiss(astdt,trtsdt) = 0 then astday = astdt-trtsdt + 1; 

run; 

 

***Assign apuper; 

%apuper(dsin1=su,dsout1=su) 

 

data su_p1; 

 length param $100 paramcd $8 parcat3 $100; 

 set su; 

 

 parcat3 = "DAILY PRODUCT USE"; 

 parcat3n = 1; 

 

 if upcase(sutrt) = "INHALER" then do; 

  param = "Daily Inhaler"; 

  paramcd = "DINHALER"; 

  paramn = 20; 

 end; 

 else if upcase(sutrt) = "NASPRAY" then do; 

  param = "Daily Nasal Spray"; 



  paramcd = "DNASPRAY"; 

  paramn = 21; 

 end; 

 else if upcase(sutrt) = "GUMS" then do; 

  param = "Daily Gums"; 

  paramcd = "DGUMS"; 

  paramn = 22; 

 end; 

 else if upcase(sutrt) = "LOZENGES" then do; 

  param = "Daily Lozenges"; 

  paramcd = "DLOZENGE"; 

  paramn = 23; 

 end; 

 else if upcase(sutrt) = "PATCHES" then do; 

  param = "Daily Patches"; 

  paramcd = "DPATCHES"; 

  paramn = 24; 

 end; 

 else if upcase(sutrt) = "OTHNRT" then do; 

  param = "Daily Other NRT Products"; 

  paramcd = "DOTHNRT"; 

  paramn = 25; 

 end; 

 else if upcase(sutrt) = "CHWSMKLS" then do; 

  param = "Daily Chewable/Smokeless Tobacco"; 



  paramcd = "DCHWMKL"; 

  paramn = 26; 

 end; 

 else if upcase(sutrt) = "CIGARS" then do; 

  param = "Daily Cigars/Cigarillos"; 

  paramcd = "DCIGARS"; 

  paramn = 27; 

 end; 

 else if upcase(sutrt) = "PIPE" then do; 

  param = "Daily Pipes"; 

  paramcd = "DPIPE"; 

  paramn = 28; 

 end; 

 else if upcase(sutrt) = "OTHTOB" then do; 

  param = "Daily Tobacco Not Listed Previously"; 

  paramcd = "DOTHTOB"; 

  paramn = 29; 

 end; 

 else if upcase(sutrt) = "E_CIG" then do; 

  param = "Daily Electronic Cigarette"; 

  paramcd = "DE_CIG"; 

  paramn = 30; 

 end; 

run; 

 



%cal_1(dsin2=su_p1,dsout2=su_p2,apuper=1) 

 

data su_p2; 

 length param $100 paramcd $8 parcat3 $100 paramtyp $10; 

 set su_p2(rename=(paramn=paramn1)); 

 

 if paramn1 = 20 then do; 

  param = "Average Daily Inhaler by Period"; 

  paramcd = "PDINHAL"; 

  paramn = 50; 

 end; 

 if paramn1 = 21 then do; 

  param = "Average Daily Nasal Spray by Period"; 

  paramcd = "PDNASPR"; 

  paramn = 51; 

 end; 

 if paramn1 = 22then do; 

  param = "Average Daily Gums by Period"; 

  paramcd = "PDGUMS"; 

  paramn = 52; 

 end; 

 if paramn1 = 23 then do; 

  param = "Average Daily Lozenges by Period"; 

  paramcd = "PDLOZENG"; 

  paramn = 53; 



 end; 

 if paramn1 = 24 then do; 

  param = "Average Daily Patches by Period"; 

  paramcd = "PDPATCHE"; 

  paramn = 54; 

 end; 

 if paramn1 = 25 then do; 

  param = "Average Daily Other NRT by Period"; 

  paramcd = "PDOTHNRT"; 

  paramn = 55; 

 end; 

 if paramn1 = 26 then do; 

  param = "Ave. Daily Chew/Smokeless Tob. by Period"; 

  paramcd = "PDCHWMKL"; 

  paramn = 56; 

 end;  

 if paramn1 = 27 then do; 

  param = "Average Daily Cigar/Cigarillo by Period"; 

  paramcd = "PDCIGARS"; 

  paramn = 57; 

 end; 

 if paramn1 = 28 then do; 

  param = "Average Daily Pipes by Period"; 

  paramcd = "PDPIPE"; 

  paramn = 58; 



 end; 

 if paramn1 = 29 then do; 

  param = "Average Daily Tob. Not Listed by Period"; 

  paramcd = "PDOTHTOB"; 

  paramn = 59; 

 end; 

 if paramn1 = 30 then do; 

  param = "Average Daily E-Cigarette by Period"; 

  paramcd = "PDE_CIG"; 

  paramn = 60; 

 end; 

 parcat3 = "AVERAGE DAILY PRODUCT USE BY PERIOD"; 

 parcat3n = 2; 

 paramtyp = "DERIVED"; 

 dtype = "AVERAGE"; 

 keep usubjid aval apuper param paramcd paramn parcat3 parcat3n paramtyp dtype parcat1 ; 

run; 

 

%cal_1(dsin2=su_p1,dsout2=su_p3,apuper=) 

 

data su_p3; 

 length param $100 paramcd $8 parcat3 $100 paramtyp $10; 

 set su_p3(rename=(paramn=paramn1)); 

 if paramn1 = 20 then do; 

  param = "Average Daily Inhaler in Ambulatory"; 



  paramcd = "ADINHAL"; 

  paramn = 70; 

 end; 

 if paramn1 = 21 then do; 

  param = "Average Daily Nasal Spray in Ambulatory"; 

  paramcd = "ADNASPR"; 

  paramn = 71; 

 end; 

 if paramn1 = 22then do; 

  param = "Average Daily Gums in Ambulatory"; 

  paramcd = "ADGUMS"; 

  paramn = 72; 

 end; 

 if paramn1 = 23 then do; 

  param = "Average Daily Lozenges in Ambulatory"; 

  paramcd = "ADLOZENG"; 

  paramn = 73; 

 end; 

 if paramn1 = 24 then do; 

  param = "Average Daily Patches in Ambulatory"; 

  paramcd = "ADPATCHE"; 

  paramn = 74; 

 end; 

 if paramn1 = 25 then do; 

  param = "Average Daily Other NRT in Ambulatory"; 



  paramcd = "ADOTHNRT"; 

  paramn = 75; 

 end; 

 if paramn1 = 26 then do; 

  param = "Ave. Daily Chew/Smokeless Tob. in Ambulatory"; 

  paramcd = "ADCHWMKL"; 

  paramn = 76; 

 end;  

 if paramn1 = 27 then do; 

  param = "Average Daily Cigar/Cigarillo in Ambulatory"; 

  paramcd = "ADCIGARS"; 

  paramn = 77; 

 end; 

 if paramn1 = 28 then do; 

  param = "Average Daily Pipes in Ambulatory"; 

  paramcd = "ADPIPE"; 

  paramn = 78; 

 end; 

 if paramn1 = 29 then do; 

  param = "Average Daily Tob. Not Listed in Ambulatory"; 

  paramcd = "ADOTHTOB"; 

  paramn = 79; 

 end; 

 if paramn1 = 30 then do; 

  param = "Average Daily E-Cigarette in Ambulatory"; 



  paramcd = "ADE_CIG"; 

  paramn = 80; 

 end;  

 parcat3 = "AVERAGE DAILY PRODUCT USE IN AMBULATORY"; 

 parcat3n = 3; 

 paramtyp = "DERIVED"; 

 dtype = "AVERAGE"; 

 keep usubjid aval param paramcd paramn parcat3 parcat3n paramtyp dtype parcat1 ; 

run; 

 

***Combine all parameter together; 

data su_p; 

 set su_p1 su_p2 su_p3; 

run; 

 

data su_p; 

 length visit $27; 

 set su_p(rename=(visit=visit1)); 

 visit = visit1; 

run; 

 

*******************************************************************************; 

 

************Step4: Prepare and output dataset                           *******;     

 



*******************************************************************************; 

 

***Combine all data together; 

data adex1; 

 length aperiodc $10 apuperc $40; 

 set dx_p ex_p su_p; 

 

 if upcase(parcat1) = "TOBACCO HEATING SYSTEM MENTHOL" then parcat1n = 1; 

 else if upcase(parcat1) = "MENTHOL CONVENTIONAL CIGARETTES" then parcat1n = 2; 

 else if upcase(parcat1) = "NRT_USE" then parcat1n = 3; 

 else if upcase(parcat1) = "TOB_USE" then parcat1n = 4; 

 

 if upcase(parcat2) = "PRODUCT USE CONFINEMENT" then parcat2n = 1; 

 else if upcase(parcat2) = "PRODUCT USE DIARY - ELECTRONIC" then parcat2n = 2; 

 else if upcase(parcat2) = "PRODUCT USE DIARY - PAPER" then parcat2n = 3; 

 

 aperiod = 1; 

 aperiodc = "Period 1"; 

 

 if apuper = 0 then apuperc = "Pre-Randomization Period"; 

 else if apuper = 1 then apuperc = "Period 1"; 

 else if apuper = 2 then apuperc = "Period 2"; 

 else if apuper = 3 then apuperc = "Period 3"; 

 else if apuper = 4 then apuperc = "Period 4"; 

 



run; 

 

***Assign analysis flag - anl01fl; 

proc sort data=adex1 out=adex1_p;where APUPER >=2 and PARCAT3 = "DAILY PRODUCT USE" and avisit 
= ''; by usubjid paramcd apuper descending aval srcseq;run; 

data adex1_p; 

 set adex1_p; 

 by usubjid paramcd apuper descending aval srcseq; 

 if first.apuper then anl01fl = 'Y'; 

run; 

 

***Assign analysis flag - anl02fl; 

proc sort data=adex1_p;by usubjid paramcd descending aval srcseq;run; 

data adex1_p; 

 set adex1_p; 

 by usubjid paramcd descending aval srcseq; 

 if first.paramcd then anl02fl = 'Y'; 

run; 

 

data adex2; 

 set adex1(where=(apuper < 2 or (apuper >=2 and ((parcat3 ne "DAILY PRODUCT USE") or 
(parcat3 = 'DAILY PRODUCT USE' and avisit ne ''))))) adex1_p; 

run; 

 

***Get data from suppdx and suppex domain; 

data suppdx; 



 length other_dx $20; 

 set sdtm.suppdx(where=(qnam = "DXOTH")); 

 other_dx = qval; 

 dxseq = input(idvarval,best.); 

 keep usubjid other_dx dxseq; 

run; 

data suppex; 

 length other_ex $20; 

 set sdtm.suppex(where=(qnam="OTHER")); 

 other_ex = qval; 

 exseq = input(idvarval,best.); 

 keep usubjid other_ex exseq; 

run; 

 

proc sort data=suppdx;by usubjid dxseq;run; 

proc sort data=adex2;by usubjid dxseq;run; 

data adex2; 

 merge adex2(in=a) suppdx; 

 by usubjid dxseq; 

 if a; 

run; 

 

proc sort data=suppex;by usubjid exseq;run; 

proc sort data=adex2;by usubjid exseq;run; 

data adex2; 



 merge adex2(in=a) suppex; 

 by usubjid exseq; 

 if a; 

run; 

 

data adex2; 

 length other $20; 

 set adex2; 

 if other_dx ne '' then other = other_dx; 

 if other_ex ne '' then other = other_ex; 

run; 

 

proc sort data=adex2;by usubjid;run; 

proc sort data=adsl;by usubjid;run; 

data adex; 

 length trtp trta $40; 

 merge adex2(in=a drop=studyid randdt trtsdt) adsl(in=b); 

 by usubjid; 

 if a; 

 trtp = trt01p; 

 trtpn = trt01pn; 

 trta = trt01a; 

 trtan = trt01an; 

 avisit = propcase(avisit); 

 if trtpn in (97 98) then do; 



  apuper = .; 

  apuperc = ''; 

 end; 

run; 

 

***Add variable attribute; 

%m_attrib_adam(dset=adex); 

 

proc sort data=adex; 

by USUBJID AVISITN PARAMCD APUPER ASTDTC AENDTC SRCDOM SRCSEQ ; 

run; 

 

 *** Output dataset; 

  data adam.adex(label = "Exposure Analysis Dataset" ); 

    set adex; 

  run; 

 

 

proc printto ;                                                                                                               

run;   

 

%m_logchk; 

 

 

 



 

 


