
/*===================================================================================

==============* 

| Covance Study Number    : 000000106331                                                           | 

| Program Name            : d_ADMH.sas                                                             | 

| Purpose                 : Create ADMH dataset                                                    |       

| Input Data              : SDTM.MH SDTM.SV ADAM.ADSL                                              |                                                   

| Output Data             : ADAM.ADMH                                                              |                                                                                                                                                                 

| Macros Called           : m_printto, m_logchk, m_attrib_adam                                                     | 

| Originally Performed by : kpothuri                                                               | 

| Date                    : 25March2015                                                            |                                                                 

|                                                                                                  | 

|====================================================================================

==============| 

| Modification History                                                                             | 

|--------------------------------------------------------------------------------------------------| 

| Modified by          :                                                                           | 

| Modification Date    :                                                                           | 

| Modification Description  :                                                                      | 

+====================================================================================

==============*/ 

options validvarname=upcase; 

 

libname adam "&base2/datasets/adam/cleaned_adam";  

 

libname sdtm "/cvn/projects/prj/data/000000106331/datasets/sdtm/sdtmx"; 

 

proc datasets lib=work nolist memtype=data kill; quit; 



 

%m_printto(route=YES); 

 

****************; 

* bring in ADSL, SV ; 

****************; 

data adsl; 

 set adam.adsl; 

 drop studyid; 

run; 

 

data SV; 

 set sdtm.sv; 

 where visitnum=1; 

 SCRNDT=input(svstdtc,yymmdd10.); 

  format SCRNDT date9.; 

 keep usubjid SCRNDT; 

run; 

 

proc sort data=adsl; by usubjid; run; 

proc sort data=sv; by usubjid; run; 

data adsl_sv; 

 merge adsl(in=a) sv; 

 by usubjid; 

 if a; 



run; 

 

*********************; 

* pick up MH      ; 

*********************; 

data mh1; 

 set sdtm.mh; 

 /*ASTDT*/ 

 if length(MHSTDTC)=10 then ASTDT=input(MHSTDTC,yymmdd10.); 

  else if length(MHSTDTC)=7 then ASTDT=.; 

  else if length(MHSTDTC)=4 then ASTDT=.; 

 format ASTDT date9.; 

 

 /*AENDT*/ 

 if length(MHENDTC)=10 then AENDT=input(MHENDTC,yymmdd10.); 

  else if length(MHENDTC)=7 then AENDT=.; 

  else if length(MHENDTC)=4 then AENDT=.; 

 format AENDT date9.; 

 

 /*MHONGFL*/ 

 if MHENRTPT = "ONGOING" then MHONGFL = "Y"; 

  else MHONGFL = "N"; 

 

 /*ANYCDFL*/ 

 if MHCAT = "CONCOMITANT DISEASE" then ANYCDFL = "Y"; 



  else ANYCDFL = "N"; 

 

 /*ANYMHFL*/  

 if MHCAT = "MEDICAL HISTORY" then ANYMHFL = "Y"; 

  else ANYMHFL = "N"; 

run; 

 

***************************; 

* Combine ADSL and EG data *; 

***************************; 

data final; 

 merge adsl_sv mh1(in=a); 

 by usubjid; 

 if a;  

  

 /*ASTDAY*/ 

 if not missing (ASTDT) and not missing (TRTSDT) then ASTDAY = ASTDT - TRTSDT + 1; 

 

 /*AENDAY*/ 

 if not missing (AENDT) and not missing (TRTSDT) then AENDAY = AENDT - TRTSDT + 1; 

 

 /*TRT:*/ 

 TRTP=TRT01P; 

 TRTPN=TRT01PN; 

 TRTA=TRT01A; 



 TRTAN=TRT01AN; 

run; 

 

************************; 

* create output dataset ; 

************************; 

data ADMH; 

 set final; 

run; 

 

%m_attrib_adam(dset=ADMH); 

 

proc sort data=ADMH out=adam.ADMH(label = 'Medical History Analysis Dataset'); 

    by usubjid mhcat mhbodsys mhdecod mhterm mhstdtc; 

run; 

 

%m_logchk; 

 

 


