
/*------------------------------------------------------------------------------ 

Covance Study ID        : COV-000000106343 

Program Name            : d_2adpc.sas    

Purpose                 : Program to create ADQSPA dataset 

Author                  : cvn_pshe 

Date of Creation        : 3MAY2015 

 

Input Data              : ADAM.ADSL,SDTM.PC,SDTM.EX,SDTM.SV 

Output Data             : ADAM.ADQSPA  

Macros Called           : m_printto,, m_logchk, m_attrib_adam   

 

-------------------------------------------------------------------------------- 

Modification History 

-------------------------------------------------------------------------------- 

Modified by             :  

Modification Date       :  

Modification Description:  

-------------------------------------------------------------------------------*/ 

options notes nosource; 

proc datasets lib=work nolist memtype=data kill; quit; 

* macro to save output and log to appropriate areas ; 

%m_printto; 

options notes source source2 nofullstimer validvarname=upcase missing=' ' mprint symbolgen; 

ods _all_ close; 

ods listing; 



 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

* pull in pkmerge data; 

* for some studies this will contain only a subset of the data please check SAP; 

 

libname adam "&base2/datasets/adam/cleaned_adam";  

libname sdtm "/cvn/projects/prj/data/000000106343/datasets/sdtm/sdtmx"; 

 

data pkmerge; 

    set adam.pkmerge;  

    format avalu $20.;  

 

    *test date and time; 

    format adt date9. atm time5. adtm datetime13.; 

    if not missing(pcdtc) and length(pcdtc) >=10 then adt=input(scan(pcdtc,1,'T'),yymmdd10.); 

    atm=input(scan(pcdtc,2,'T'),time8.); 

    if nmiss(adt,atm) =0 then adtm=dhms(adt, 0,0,atm); 

    if avisit='Day 6' then avisit='Day 6/Discharge'; 

    avalu=trim(units); 

    drop siteid weightbl exdose exdosu trta: trtp: aday pcorres ; 

run; 

 

%m_chglength(inds=pkmerge,varlist=avalc, lenlist=$200); 



 

* further code will be required to include data from SDTM.PC for other studies; 

* identify SA arm subjects; 

 

data sa; 

    set adam.adsl(where=(armcd='SMABST')); 

    keep usubjid; 

run; 

    

*pick up data not used in PKmerge; 

data pc; 

    merge sdtm.pc(in=b) sa(in=a); 

    by usubjid; 

    if a then sa=1; 

    if b; 

 

    format avalc $200. aval best. avalu $20. param $120. paramcd $3. paramn subjidn 8.;  

    subjidn=input(scan(usubjid,6,'-'),best.); 

 

    *results; 

    aval=pcstresn; 

    avalc=pcstresc; 

    avalu=pcstresu; 

 

    *parameter category; 



    if pctestcd ne '' then do; 

      if not missing(pcstresu) and pctest ne '' then param=trim(pctest) ||' ('||trim(pcstresu)||')';  

      else if pctestcd='COT' then param='Cotinine (ng/mL)'; 

      else if  pctestcd='NIC' then param='Nicotine (ng/mL)'; 

      paramcd=pctestcd; 

    end; 

 

    if pctestcd='NIC' then 

        paramn=1; 

    else if pctestcd='COT' then 

        paramn=2; 

    else put 'PCTEST not recognised : ' pctestcd=; 

 

    *treatment period; 

    format aperiod 8.; 

    aperiod=1; 

  

    format adt date9. atm time5. adtm datetime13.; 

    if not missing(pcdtc) and length(pcdtc) >=10 then adt=input(scan(pcdtc,1,'T'),yymmdd10.); 

    atm=input(scan(pcdtc,2,'T'),time8.); 

    if nmiss(adt,atm) =0 then adtm=dhms(adt, 0,0,atm); 

 

    *visits and timepoints; 

    format avisit atpt $40. avisitn atptn 8.; 

    avisit=propcase(visit); 



    avisitn=visitnum; 

     

    if avisit in ('Day 0' 'Day 1' 'Day 2' 'Day 3' 'Day 4') then do; 

       atpt='08:00 PM - 09:30 PM'; 

       atptn=11; 

       end; 

 

    if avisit in ('Day 30' 'Day 60') then do; 

       atpt='10:00 AM - 11:30 AM'; 

       atptn=12; 

       end; 

 

    if avisit in ('Day 90') then do; 

       atpt='10:00 AM - 12:30 PM'; 

       atptn=13; 

       end; 

 

    if sa=1 and avisit ='Day 5' then do; atpt='08:00 PM - 09:30 PM'; 
atptn=11; end;  

    else if sa=1 and avisit ='Day 6/Discharge Confinement' then do; 
atpt='08:00 AM - 09:30 AM'; atptn=14; end;  

 

    keep usubjid subjidn pcseq pcspec pcstat pcreasnd param: avisit: atpt: aperiod aval avalc avalu pcdtc 
adt atm adtm   

        pcorres pclloq sa;   

 

    if a then output;  * SA data not included in pkmerge; 



    else if not a and not(avisit in ('Day 5', 'Day 6/Discharge Confinement')) then output; * all other data 
not in Day 5 and Day 6;  

run; 

 

* set all and populated PCLLOQ***;  

data adpc; 

    set pkmerge pc;   

    format dtype $20.  ;  

    rename ntime = pnomtime; 

    format ablfl $2.;   

    if avisit='Day 0' and aval ne . then  ablfl='Y'; 

    if index(avalc,'BLQ') then do; 

      aval=.; 

      dtype=' '; 

      bloqfl='Y'; 

      pclloq = input(scan(avalc, 2, '< ( )'), best.); 

    end; 

run; 

 

* derive bloqfl and imputted records; 

data blq; 

    set adpc; 

    format bloqfl $2.; 

    if index(avalc,'BLQ'); 

     aval=pclloq/2;  

     avalc=strip(put(aval, best.)); 



     dtype='BLQHALF'; 

     bloqfl='Y'; 

run; 

 

data adpc;  

   set adpc blq; 

run; 

 

proc sort data=adpc;  

    by usubjid subjidn pcseq; 

run; 

 

* bring in pc data for day ; 

data pca; 

    set sdtm.pc; 

    format subjidn 8.; 

    subjidn=input(scan(usubjid,6,'-'),best.); 

    keep usubjid pcseq pcdy subjidn epoch; 

run; 

 

proc sort data=pca; 

    by usubjid subjidn pcseq; 

run; 

 

data adpc; 



    merge pca adpc (in=a);   

    by usubjid subjidn pcseq; 

    if a; 

run; 

 

*bring in screening date; 

data sv; 

    set sdtm.sv(where=(visit='SCREENING')); 

    format scrndt date9.; 

    scrndt=input(svstdtc,yymmdd10.); 

    format subjidn 8.; 

    subjidn=input(scan(usubjid,6,'-'),best.); 

    keep usubjid scrndt subjidn; 

run; 

 

proc sort data=adam.adsl out=adsl; 

   by usubjid subjidn; 

run;  

 

proc sort data=sv; 

   by usubjid subjidn; 

run; 

 

* bring in adsl for T0 and subject level data; 

data adsl; 



    merge adsl adpc(in=a) sv; 

    by usubjid subjidn; 

    if a; 

 

    format aperiodc $8.;  

    aperiodc='Period '||compress(put(aperiod,1.)); 

 

    format trta trtp $40. trtan trtpn 8.; 

 

    if aperiod=1 then do; 

        trta=trt01a; 

        trtp=trt01p; 

        trtan=trt01an; 

        trtpn=trt01pn; 

    end; 

run; 

 

*need first product use dates and times; 

*included all so that there is needed; 

data ex; 

    set adam.adex; 

    if astday ge 0; 

    keep usubjid astdt astdtm astday; 

run; 

 



proc sort data=ex; 

    where astdtm ne .; 

    by usubjid astday astdtm; 

run; 

 

data ex; 

    set ex; 

    by usubjid astday; 

    if first.astday; 

run; 

 

proc transpose data=ex out=texdx prefix=tr; 

    var astdtm; 

    by usubjid; 

    id astday; 

run; 

 

* windows; 

data adpc2; 

    format awlo awhi datetime13. awrange $50. aday 8.; ;  

    merge adsl(in=a) texdx; 

    by usubjid; 

    if a; 

    if nmiss(adt, trtsdt)=0 then aday=adt-trtsdt+1; 

 



    if trta in ('THSm2.2', 'mCC') and not missing(tr5) then trtstart=tr5; 
/*specific for REXC where pkday = day 5*/  

 

    if nmiss(adt)=0 then do; 

    if avisit in ('Day 0', 'Day 1', 'Day 2', 'Day 3', 'Day 4') then do; 

       awlo=adt*24*60*60+'20:00't; 

       awhi=adt*24*60*60+'21:30't; 

    end; 

    else if avisit in ('Day 30', 'Day 60')  then do; 

       awlo=adt*24*60*60+'10:00't; 

       awhi=adt*24*60*60+'11:30't; 

    end; 

    else if avisit in ('Day 90')  then do; 

       awlo=adt*24*60*60+'10:00't; 

       awhi=adt*24*60*60+'12:30't; 

    end; 

    end; 

 

    if avisit in ('Day 5', 'Day 6/Discharge Confinement') and trta in 
('THSm2.2', 'mCC') and nmiss(trtstart)=0 then do;  

      if atptn=0 then do; 

        awlo=trtstart-'00:15't; 

        awhi=trtstart; 

      end;  

      else if atptn=1 then do; /* 2 h */ 

        awlo=trtstart+'02:00't; 



        awhi=trtstart+'02:05't; 

      end;  

      else if atptn=2 then do; /* 4 h */ 

        awlo=trtstart+'04:00't; 

        awhi=trtstart+'04:05't; 

    end;  

    else if atptn=3 then do; /* 6 h */ 

        awlo=trtstart+'06:00't; 

        awhi=trtstart+'06:05't; 

    end;  

    else if atptn=4 then do; /* 8 h */ 

        awlo=trtstart+'08:00't; 

        awhi=trtstart+'08:05't; 

    end;  

    else if atptn=5 then do; /* 10 h */ 

        awlo=trtstart+'10:00't; 

        awhi=trtstart+'10:05't; 

    end;  

    else if atptn=6 then do; /* 12 h */ 

        awlo=trtstart+'12:00't; 

        awhi=trtstart+'12:05't; 

    end;  

    else if atptn=7 then do; /* 14 h */ 

        awlo=trtstart+'14:00't; 

        awhi=trtstart+'14:05't; 



    end;  

    else if atptn=8 then do; /* 16 h */ 

        awlo=trtstart+'16:00't; 

        awhi=trtstart+'16:05't; 

    end;  

    else if atptn=9 then do; /* 20 h */ 

        awlo=trtstart+'20:00't; 

        awhi=trtstart+'20:05't; 

    end;  

    else if atptn=10 then do; /* 24 h */ 

        awlo=trtstart+'24:00't; 

        awhi=trtstart+'24:05't; 

    end; 

 

    end; 

   

    if trta in ('SA') and nmiss(adt)=0 then do;  

        if avisit in ('Day 5') then do;  

        awlo=adt*24*60*60+'20:00't; 

        awhi=adt*24*60*60+'21:30't; 

        end; 

        else if avisit in ('Day 6/Discharge Confinement') then do ; 

        awlo=adt*24*60*60+'08:00't; 

        awhi=adt*24*60*60+'09:30't; 

        end; 



    end; 

   

    if nmiss(awlo, awhi)=0  then awrange=strip(put(awlo,datetime16.)) ||'-
'||strip(put(awhi,datetime16.)); 

run; 

 

data adpc2; 

   set adpc2; 

   format devn devw apuper 8.  DEVWC apuperc $10.; 

 

     if nmiss(PACTIME, PNOMTIME)=0 then  DEVN=PACTIME-PNOMTIME;   

 

  if .<ADTM<AWLO then DEVW=(ADTM-AWLO)/60 ; 

     else if ADTM>AWHI>. then DEVW=(ADTM-AWHI)/60 ; 

  if devw ne . then do; 

    devwc = strip(put(round(devw,0.001),best8.3)) || " min"; 

    if devw gt 0 then devwc = cats("+",devwc); 

  end; 

 

   if nmiss(DEVN, PNOMTIME)=0 and PNOMTIME ne 0 then 
PCTDEV=ROUND(100*DEVN/PNOMTIME, 0.1); 

   

   if 101<=avisitn<=106 then apuper=1; 

     else If 106<avisitn<=131 then apuper=2; 

     else if 131<avisitn<=161 then apuper=3; 

     else if 161<avisitn<=191 then apuper=4; 



 

   if apuper=1 then apuperc= 'Period 1'; 

     else if apuper=2 then apuperc='Period 2'; 

     else if apuper=3 then apuperc='Period 3'; 

     else if apuper=4 then apuperc='Period 4'; 

 

   if TRTPN=97 | TRTPN=98 then do; APUPER=.; APUPERC = ''; end; 

 

run; 

 

* Analysis records flag; 

data adpc2; 

   set adpc2; 

   format anl01fl $2. aexreas $200.; 

    

   if  bloqfl='' and aval ne . then do; 

     if avisit in ('Day 0','Day 1','Day 2','Day 3','Day 4', 'Day 5', 'Day 6/Discharge 
Confinement' ) then do; 

     if  nmiss(awlo, awhi, adtm)=0 and awlo <= adtm <= awhi then anl01fl='Y'; 

     end; 

     else if avisit not in ('Day 0','Day 1','Day 2','Day 3','Day 4', 'Day 5', 
'Day 6/Discharge Confinement' ) then anl01fl='Y'; 

    end; 

   else if bloqfl='Y' and aval ne . and dtype = "BLQHALF" then anl01fl='Y';    

      

   if compress(fasfl)='N' then aexreas=strip(fasreas); 



   if anl01fl ='' then aexreas=trim(fasreas); 

     

run; 

 

data adpc2; 

   set adpc2; 

   if not missing(atm) then tm8_dist=abs(atm-'20:00't) ; 

 

proc sort data=adpc2; 

   by usubjid paramn avisit dtype tm8_dist ; 

run ; 

 

data adpc2; 

   set adpc2; 

   format anl02fl $2.; 

   by usubjid paramn avisit dtype tm8_dist ; 

   if first.avisit and aval^=. and upcase(avisit) in ('DAY 0' 'DAY 1' 'DAY 2' 'DAY 3' 'DAY 4' 'DAY 5') then 
anl02fl='Y' ; 

run; 

 

*change from baseline; 

proc sort data=adpc2; 

    by usubjid subjidn paramn avisitn atptn; 

run; 

 

data base(rename=(aval=base)); 



    set adpc2 (where=(ablfl='Y')); 

    keep usubjid subjidn paramn aval; 

run; 

 

proc sort data=base ; 

   by usubjid subjidn paramn; 

run; 

 

* here delete one baseline record from the double baslines in sdtm.pc; 

data base; 

   set base; 

    by usubjid subjidn paramn; 

    if not first.paramn then delete; 

run; 

 

data change; 

    merge adpc2 base;  

    by usubjid subjidn paramn; 

    format chg pchg best. pchgc $20.; 

    LENGTH PCREASN2 $200 PCDTC2 $16;  

    if nmiss(aval, base)=0 then do; 

    chg=aval-base; 

    pchg=(aval-base)/base*100; 

    pchgc=STRIP(put(round(pchg,0.1),8.1)); 

    end;  



 

    PCREASN2=PCREASND;   

    PCDTC2=PCDTC;  

 

    if subjid='TOK-0146' and ablfl='Y' and aval=33 then ablfl=''; 

 

    DROP PCREASND PCDTC;  

    RENAME PCREASN2=PCREASND PCDTC2=PCDTC; 

run; 

 

************************; 

* create output dataset ; 

************************; 

options replace; 

 

proc sort data = change out=adpc; 

    BY USUBJID AVISITN PARAMCD ATPTN;  

run; 

 

%m_chglength(inds=adpc,varlist=PARAMCD AVALC AVALU PCREASND, lenlist= $3 $11 $5 $41); 

 

%m_attrib_adam (dset=ADPC); 

 

data adpc; 

   set adpc; 



   format _all_; 

   format SCRNDT ADT TRTSDT TRTEDT DATE9. AWLO AWHI TRTSDTM TRTEDTM 
ADTM DATETIME13. ATM time5.; 

run; 

 

data  adam.adpc (label= 'Pharmacokinetic Concentration Analysis Dataset'); 

  set adpc; 

run; 

 

options noreplace; 

 

proc printto; run; 

 

%m_logchk; 

 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 

 

 

 

 

     

 


