
* 

* 

Population 

* 

* Date        Author      Ref    Revision (Date in YYYYMMDD format)  

**********************************************************************; 

 

%let prgname=T15020105_ZRHM_REXA_07_JP_V1; 

options nomprint nosymbolgen nomlogic validvarname=upcase; 

options sasautos=("W:\pmp07\macros" sasautos) notes; 

%init(delivery=9); 

%titlecsv(prgname=&prgname.); 

%put &title1; 

%put &title2; 

%put &APPENDIX; 



%put &endpoint; 

%put &outname.; 

 

data adsl; 

    set adam.adsl(where=(safbfl='Y')); 

    if      trt01an=4  then do; trt=1; output; end; 

    else if trt01an=5  then do; trt=2; output; end; 

    else if trt01an=3  then do; trt=3; output; end; 

    else if trt01an=96 then do; trt=4; output; end; 

    trt=5; output; 

run ; 

 

proc freq data =adsl noprint;  

    table trt/out=treatabt (rename =(count=total)drop=percent); 

run; 

 

data _null_;        

    Set treatabt; 

    Call symput('n'||strip(put(trt, best.)),strip(put(total, best.)));     

Run;   

%put &n1 &n2 &n3 &n4 &n5; 

 

*** Prepare data for analysis ***; 

proc sql; 

    create table data0 as 



    select a.trt, b.* 

    from adsl as a cross join adam.adfa as b 

    where a.usubjid=b.usubjid and b.paramcd in ('NYIELD' 'TYIELD') and avisit='Screening' 

    ; 

quit; 

 

proc sort data=data0; by usubjid trt brand paramcd ; run; 

 

data data1; 

    set data0; 

    by usubjid trt brand paramcd ; 

    retain cat1 cat2; 

    if      paramcd='NYIELD' then cat1=aval; 

    else if paramcd='TYIELD' then cat2=aval; 

    if last.brand then output; 

run; 

 

proc means data=data1(where=(trt=5)) noprint nway; 

    class brand trt; 

    var trt; 

    output out=four(drop=_: where=(count<4)) n=count; 

run; 

 

proc sort data=data1; by brand ; run; 

 



data data2; 

    merge data1(in=a) four(in=b keep=brand); 

    by brand; 

    if b then brand='Other'; 

run; 

 

%macro mfreq(in_dsn=, n_max=, order1=, order2=, class=); 

proc means data=&in_dsn noprint nway; 

    class &class trt; 

    var trt; 

    output out=stat1 n=count; 

run; 

 

data stat2 (drop=_type_ _freq_  percentx); 

    set stat1; 

    by &class trt; 

 

    length percentage $25; 

    %do i=1 %to &n_max; 

        if trt=&i and count>. then percentx=count/%eval(&&n&i)*100; 

        if      percentx=100  then percentage=put(count,4.)||' (100)'; 

        else if percentx>=0.1 then percentage=put(count,4.)||' ('||strip(put(percentx,5.1))||')'; 

        else if percentx>.    then percentage=put(count,4.)||' (<0.1)'; 

    %end; 

run; 



 

proc transpose data=stat2 out=stat3 prefix=col; 

    by &class; 

    id trt; 

    var percentage; 

run; 

 

data final_&order1 (drop=_name_); 

    length col1-col&n_max. $100; 

    set stat3; 

    order2=&order2; 

    order1=&order1; 

    %do i=1 %to &n_max; 

        if col&i='' then col&i='0'; 

    %end; 

run; 

%mend mfreq; 

 

%mfreq(in_dsn=data2, n_max=5, order1=1, order2=1, class=brand cat1 cat2); 

 

*** Prepare the output data set per mock-up/shell ***; 

data final1; 

    set final_1; 

    brand=propcase(brand); 

    if brand='Other' then order1=2; 



    order2=input(scan(col5,1,'('),best.); 

run; 

 

proc sort; by order1 descending order2; run; 

 

data final1; 

    set final1; 

    pageno=1; 

    if _n_>16 then pageno=2; 

run; 

 

data odata.%sysfunc(scan(&prgname,1,'_')); 

    set final1(in=a); 

    if a then group="Part1"; 

run; 

 

%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.)); 

 

%global totalpage; 

 

data _null_; 

  set final1 end=eof; 

  if eof then do; 

    call symput('totalpage', trim(left(put(pageno,8.)))); 

  end; 



run; 

 

%put totalpage=&totalpage; 

 

%macro reppart; 

 

  %do i = 1 %to &totalpage; 

proc report data=final1(where=(pageno=&i.))  headskip headline spacing=4 nowd split='|' 

style=[outputwidth=100%] style(header column)=[protectspecialchars=off]; 

    column pageno order1 order2 brand cat1 cat2 col1-col5; 

 

    define pageno  /order order noprint; 

    define order1  /order order noprint; 

    define order2  /order descending order noprint; 

    define brand   /display "Brand"  style(column)=[cellwidth=25% asis=on] style(header)=[just=l]; 

    define cat1    /display "ISO nicotine|yield|(mg)"           flow style(column)=[cellwidth=10% just=c]; 

    define cat2    /display "ISO tar|yield|(mg)"                flow style(column)=[cellwidth=10% just=c]; 

    define col1    /display "THSm2.2|(N=&n1)|n(%)"              flow style(column)=[cellwidth=10% just=c]; 

    define col2    /display "mCC|(N=&n2)|n(%)"                  flow style(column)=[cellwidth=10% just=c]; 

    define col3    /display "SA|(N=&n3)|n(%)"                   flow style(column)=[cellwidth=10% just=c]; 

    define col4    /display "Product|Test|(N=&n4)|n(%)"         flow style(column)=[cellwidth=10% just=c]; 

    define col5    /display "Overall|Safety|(N=&n5)|n(%)"       flow style(column)=[cellwidth=10% just=c]; 

 

    compute before pageno ; 

        line ""; 

    endcomp; 



    compute before _page_ /style=[fontweight=bold fontsize=3.75]; 

        line @1 "&title1 &title2"; 

        line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  "; 

    endcomp; 

    compute after _page_/style=[fontsize=1.75]; 

        line @1 "Note: �Product Test� refers to all subjects who tested the THS product but were not 

randomized."; 

        line @1 "Note: The Overall Safety refers to all subjects in the Safety Population."; 

        line @1 "Note: mCC = Menthol conventional cigarettes; SA = Smoking abstinence; THSm2.2 = 

Tobacco Heating System 2.2 Menthol."; 

        line @1 "Note: Percentages are based on the number of subjects indicated in the column header 

(N)."; 

        line @1 "Note: Only brands used by at least 4 subejcts in the Safety Population are presented."; 

        line @1 ""; 

        line @1 "&APPENDIX."; 

        line @1 "Study ID:ZRHM-REXA-07-JP         Program: &fprgname..sas        Status: 

&repversion./&fdate.          Page: &i of &totalpage."; 

    endcomp; 

 

    compute after pageno ; 

        line ""; 

    endcomp; 

run; 

%end; 

 

%mend; 

%reppart; 



 

ods listing; 

ods rtf close; 

 


