
**********************************************************************; 

* Project           : ZRHM-REXA-07-JP 

* 

* Program name      : t15020107_ZRHM-REXA-07_V1.sas 

* 

* Author            : W. Yang  

* 

* Date created      : 05/21/2015 

* 

* Purpose           : Create Table 15.2.1.6 Summary of Concomitant Diseases - Safety Population 

* 

* Revision History  : 

* 

* Date        Author      Ref    Revision (Date in YYYYMMDD format)  

* 

**********************************************************************; 

 

%let prgname=T15020107_ZRHM_REXA_07_JP_V1; 

options nomprint nosymbolgen nomlogic validvarname=upcase; 

options sasautos=("W:\pmp07\macros" sasautos) notes; 

%init(delivery=9); 

%titlecsv(prgname=&prgname., version=3); 

%put &title1; 

%put &title2; 

%put &APPENDIX; 



%put &endpoint; 

%put &outname.; 

 

data adsl1; 

    set adam.adsl; 

    where safbfl='Y'; 

    if      trt01an=4 then do; trt=1; output; end; 

    else if trt01an=5 then do; trt=2; output; end; 

    else if trt01an=3 then do; trt=3; output; end; 

    if trt01an=96     then do; trt=4; output; end; 

    if safbfl='Y'   then do; trt=5; output; end; 

run ; 

 

proc freq data =adsl1 noprint;  

  table trt/out=treatabt (rename =(count=total)drop=percent); 

run; 

 

Data _null_;        

  Set treatabt; 

  Call symput('n'||strip(put(trt, best.)),strip(put(total, best.)));     

Run;   

 

%put &n1 &n2 &n3 &n4 &n5; 

 

**prepare data for macro/frequency in order;   



data admh1; 

    set adam.admh; 

    where mhcat='CONCOMITANT DISEASE' and safbfl='Y' and mhdecod^=''; 

    if      trtan=4 then do; trt=1; output; end; 

    else if trtan=5 then do; trt=2; output; end; 

    else if trtan=3 then do; trt=3; output; end; 

    if trtan=96     then do; trt=4; output; end; 

    if safbfl='Y' then do; trt=5; output; end; 

run; 

 

proc sort data=admh1 out=top0;   by usubjid trt; run; 

proc sort data=top0 out=top nodupkey;    by usubjid trt; run; 

 

proc sort data=admh1 out=body0;  by usubjid mhbodsys trt; run; 

proc sort data=body0 out=body nodupkey;   by studyid usubjid mhbodsys trt; run; 

 

proc sort data=admh1 out=term0;  by usubjid mhbodsys mhdecod trt; run; 

proc sort data=term0 out=term nodupkey;   by usubjid mhbodsys mhdecod trt; run; 

 

 

/***************************************************************** 

Categorical Statistics for cm Tables, n (%) 

******************************************************************/ 

 

%macro mfreq(in_dsn=, n_max=, order1=, order2=, class=); 



proc means data=&in_dsn noprint nway; 

    class &class trt; 

    var trt; 

    output out=stat1_a n=count; 

run; 

 

proc means data=&in_dsn.0 noprint nway; 

    class &class trt; 

    var trt; 

    output out=stat1_b n=count0; 

run; 

 

data stat2 (drop=_type_ _freq_  percentx); 

    merge stat1_a stat1_b (drop=_type_ _freq_); 

    by &class trt; 

 

    length percentage $25; 

    %do i=1 %to &n_max; 

        if trt=&i and count>. then percentx=count/%eval(&&n&i)*100; 

        if      percentx=100  then percentage=put(count,4.)||' (100)'||put(count0,3.0); 

        else if percentx>=0.1 then percentage=put(count,4.)||' 

('||strip(put(percentx,5.1))||')'||put(count0,3.0); 

        else if percentx>.    then percentage=put(count,4.)||' (<0.1)'||put(count0,3.0); 

    %end; 

run; 

 



proc transpose data=stat2 out=stat3 prefix=col; 

    by &class; 

    id trt; 

    var percentage; 

run; 

 

data final_&order1 (drop=_name_); 

    set stat3; 

    order2=&order2; 

    order1=&order1; 

    %do i=1 %to &n_max; 

        if col&i='' then col&i='0'; 

    %end; 

run; 

%mend mfreq; 

 

%mfreq(in_dsn=top,  n_max=5, order1=1, order2=1, class=); 

%mfreq(in_dsn=body, n_max=5, order1=2, order2=2, class=mhbodsys); 

%mfreq(in_dsn=term, n_max=5, order1=3, order2=2, class=mhbodsys mhdecod); 

 

data final1; 

    set final_:; 

run; 

 

data final2; 



    length stat $100; 

    set final1; 

 

    if      order1=1 and order2=1 then stat='Any Concomitant Diseases'; 

    else if order1=2 and order2=2 then stat=strip(propcase(mhbodsys)); 

    else if order1=3 and order2=2 then stat='   '||strip(propcase(mhdecod)); 

 

    order3=sum(input(scan(col5,1,'('),8.)); 

run; 

 

proc sort  data=final2; by order2 mhbodsys mhdecod descending order3 ; run; 

 

data final; 

    set final2; 

    by order2 mhbodsys mhdecod descending order3 ; 

    retain body_ord term_ord zero1-zero5; 

     

    %macro mtemp; 

        if first.mhbodsys then do; %do i=1 %to 5; if col&i.='0' then zero&i.=1; else zero&i.=0; %end; end; 

        else                   do; %do i=1 %to 5; if zero&i.=1 then col&i=''; %end; end; 

    %mend; 

    %mtemp; 

 

    if first.mhdecod  then term_ord=order3; 

    if first.mhbodsys then body_ord=order3; 



 

    if mhbodsys='' then mhbodsys='AAA'; 

run; 

 

proc sort data=final; by order2 descending body_ord mhbodsys descending term_ord mhdecod; run; 

 

data final; 

    set final; 

    pageno=1; 

    if _n_>13 then pageno=2; 

    if _n_>32 then pageno=3; 

run; 

 

%global totalpage; 

 

data _null_; 

  set final end=eof; 

  if eof then do; 

    call symput('totalpage', trim(left(put(pageno,8.)))); 

  end; 

run; 

 

%put totalpage=&totalpage; 

%let sline=\brdrb\brdrs\brdrw10\brsp20; 

 



%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.)); 

 

%macro reppart; 

%do i = 1 %to &totalpage; 

proc report data=final(where=(pageno=&i.)) headskip headline spacing=4 nowd split='|' 

style=[outputwidth=100%] style(header column)=[protectspecialchars=off]; 

    column pageno order2 mhbodsys ("^S={just=left}System Organ Class"  stat) ("THSm2.2|(N=&n1) 

&sline."  col1)  

        ("mCC|(N=&n2) &sline."  col2)  ("SA|(N=&n3) &sline."  col3)  ("Product Test|Only|(N=&n4) &sline."  

col4)  

        ("Overall|Safety|(N=&n5) &sline."  col5) ; 

 

    define pageno   /order noprint; 

    define order2   /order noprint; 

    define mhbodsys /order order=data noprint; 

    define stat    /display "^S={just=left}   Preferred Term"  style(column)=[cellwidth=36% asis=on] 

style(header)=[just=l]; 

    define col1    /display "n(%) Events"   flow style(column)=[cellwidth=12% just=c]; 

    define col2    /display "n(%) Events"   flow style(column)=[cellwidth=12% just=c]; 

    define col3    /display "n(%) Events"   flow style(column)=[cellwidth=12% just=c]; 

    define col4    /display "n(%) Events"   flow style(column)=[cellwidth=12% just=c]; 

    define col5    /display "n(%) Events"   flow style(column)=[cellwidth=12% just=c]; 

 

    compute before mhbodsys ; 

        line ""; 

    endcomp; 

    compute before _page_ /style=[fontweight=bold fontsize=3.75]; 



        line @1 "&title1 &title2"; 

        line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  "; 

    endcomp; 

    compute after _page_/style=[fontsize=1.75]; 

        line @1 'Note: "Product Test Only" refers to all subjects tested the THS product but were not 

randomized.';; 

        line @1 "Note: mCC = Menthol conventional cigarettes; SA = Smoking abstinence; THSm2.2 = 

Tobacco Heating System 2.2 Menthol."; 

        line @1 "Note: Percentages are based on the number of subjects indicated in the column header 

(N)."; 

        line @1 ""; 

        line @1 "&APPENDIX."; 

        line @1 "Study ID:ZRHM-REXA-07-JP         Program: &fprgname..sas        Status: 

&repversion./&fdate.          Page: &i of &totalpage."; 

    endcomp; 

 

    compute after pageno ; 

        line ""; 

    endcomp; 

run; 

%end; 

%mend; 

%reppart; 

 

ods listing; 

ods rtf close; 

 



data odata.%sysfunc(scan(&prgname,1,'_')); 

    set final(in=a); 

    if a then group="SAFBFL"; 

run; 

 


