
**********************************************************************; 

* Project           : ZRHM-REXA-07-JP 

* 

* Program name      : t1502020102_ZRHM-REXA-07_V1.sas 

* 

* Author            : W. Yang  

* 

* Date created      : 05/27/2015 

* 

* Purpose           : Create Table 15.2.2.1.2 Summary of Daily Product Use in Confinement Period � PP Set 

* 

* Revision History  : 

* 

* Date        Author      Ref    Revision (Date in YYYYMMDD format)  

* 

**********************************************************************; 

 

%let prgname=T1502020102_ZRHM_REXA_07_JP_V1; 

options nomprint nosymbolgen nomlogic validvarname=upcase; 

options sasautos=("W:\pmp07\macros" sasautos) notes; 

%init(delivery=9); 

%titlecsv(prgname=&prgname., version=3); 

%put &title1; 

%put &title2; 

%put &APPENDIX; 



%put &endpoint; 

%put &outname.; 

 

%let pop1fl=pprot1fl; %let sub=1; %let out=out1; 

%*macro repeat(pop1fl=, sub=, out=); 

data adsl; 

    set adam.adsl; 

    where &pop1fl.='Y'; 

    if      trt01an=4 then do; trt=1; output; end; 

    else if trt01an=5 then do; trt=2; output; end; 

    else if trt01an=3 then do; trt=3; output; end; 

    trt=4; output;  

run ; 

 

proc freq data =adsl noprint;  

  table trt/out=treatabt (rename =(count=total)drop=percent); 

run; 

 

Data _null_;        

  Set treatabt; 

  Call symput('n'||strip(put(trt, best.)),strip(put(total, best.)));     

Run;   

 

%put &n1 &n2 &n3 &n4; 

 



*** Prepare data for analysis ***; 

proc sort data=adam.adex out=adex; 

    by usubjid avisitn avisit;  

    where parcat2='PRODUCT USE CONFINEMENT' and parcat3n=1; 

run; 

 

proc sql; 

    create table data0 as 

    select a.trt, b.* 

    from adsl as a cross join adex as b 

    where a.usubjid=b.usubjid 

    ; 

quit; 

 

*** Macro for both Categorical & Continuous variables statistics ***; 

%macro mfreq(in_dsn=, n_max=, order1=, order2=, class=); 

proc means data=&in_dsn noprint nway; 

    class &class trt; 

    var trt; 

    output out=stat1 n=count; 

run; 

 

data stat2 (drop=_type_ _freq_  percentx); 

    set stat1; 

    by &class trt; 



 

    length percentage $25; 

    %do i=1 %to &n_max; 

        if trt=&i and count>. then percentx=count/%eval(&&n&i)*100; 

        if      percentx=100  then percentage=put(count,4.)||' (100)'; 

        else if percentx>=0.1 then percentage=put(count,4.)||' ('||strip(put(percentx,5.1))||')'; 

        else if percentx>.    then percentage=put(count,4.)||' (<0.1)'; 

    %end; 

run; 

 

proc transpose data=stat2 out=stat3 prefix=col; 

    by &class; 

    id trt; 

    var percentage; 

run; 

 

data final_1 (drop=_name_); 

    length col1-col&n_max. $100; 

    set stat3; 

    order2=&order2+1; 

    order1=&order1; 

    %do i=1 %to &n_max; 

        if col&i='' then col&i='0'; 

    %end; 

run; 



%mend mfreq; 

 

%mfreq(in_dsn=data0(where=(avisitn=98 and paramn=1)), n_max=4, order1=1, order2=aval, 

    class=avisitn avisit paramn param aval); 

 

%macro mmeans(in_dsn=, class=, var=, order1=); 

proc means data=&in_dsn noprint nway alpha=0.05; 

    class &class trt; 

    var &var; 

    output out= stat1 n=_n mean=_mean std=_std median=_median min=_min max=_max lclm=_lclm 

uclm=_uclm; 

run; 

 

data stat2 (drop=_:); 

    length n meansd median minmax ci $50 fm1-fm3 $10; 

    set stat1; 

    deci=0; 

    fm1=cats(12,".",strip(put(deci+1,3.))); 

    fm2=cats(12,".",strip(put(deci+2,3.))); 

    fm3=cats(12,".",strip(put(deci+0,3.))); 

 

    _std=ceil(_std*100)/100; 

    _uclm=ceil(_uclm*100)/100; 

    _lclm=floor(_lclm*100)/100; 

    if _lclm=. and _std=0 and _mean>. then do; _lclm=_mean; _uclm=_mean; end; 

 



    if      _mean ne . and _std ne . then meansd=strip(putn(_mean, fm1))||" 

("||strip(putn(_std,fm2))||")"; 

    else if _mean ne . and _std eq . then meansd=strip(putn(_mean, fm1))||" (NA)"; 

    if _lclm    ne . then ci=strip(putn(_lclm,fm3))||", "||strip(putn(_uclm,fm3)); 

    if _lclm=. and _std=0 and _mean>. then ci=strip(putn(_mean,fm3))||", "||strip(putn(_mean,fm3)); 

    if _min     ne . then minmax=strip(putn(_min,fm3))||", "||strip(putn(_max,fm3)); 

    if _n       ne . then n=strip(put(_n,best12.)); 

    if _median  ne . then median=strip(putn(_median, fm1)); 

run; 

 

proc transpose data=stat2 out=stat3 prefix=col; 

    by &class; 

    var n meansd ci median minmax; 

    id trt; 

run; 

 

data final_&order1. (drop=_name_); 

    length stat $100; 

    set stat3;   

    if _name_='N'      then do; stat="n";         order2 =1; end; 

    if _name_='MEANSD' then do; stat="Mean (SD)"; order2 =5; end; 

    if _name_='CI'     then do; stat="95% CI";    order2 =6; end; 

    if _name_='MEDIAN' then do; stat="Median";    order2 =7; end; 

    if _name_='MINMAX' then do; stat="Min, Max";  order2 =8; end; 

    order1=&order1; 

run; 



%mend mmeans; 

 

%mmeans(in_dsn=data0, class=avisitn avisit paramn param, var=aval, order1=2); 

 

data final1; 

    set final_:; 

    if avisitn=98 then order1=1; 

run; 

 

proc sort data=final1; by order1 avisitn avisit paramn order2; run; 

 

data shell; 

    length avisit $40; 

    order1=1; avisitn=98;  paramn=1; order2=1; avisit='Day -2'; occasion='THSm2.2 Test            '; output; 

    order1=1; avisitn=98;  paramn=1; order2=4; occasion=''; output; 

/*    order1=2; avisitn=99;  paramn=11; order2=1; avisit='Day -1'; occasion='Pre-Randomization Period'; 

output;*/ 

    order1=2; avisitn=99;  paramn=1; order2=1; avisit='Day -1'; occasion='Pre-Randomization Period'; 

output; 

    order1=2; avisitn=99;  paramn=1; order2=5; avisit='Day -1'; occasion=''; output; 

    order1=2; avisitn=99;  paramn=1; order2=6; avisit='Day -1'; occasion=''; output; 

    order1=2; avisitn=99;  paramn=1; order2=7; avisit='Day -1'; occasion=''; output; 

    order1=2; avisitn=99;  paramn=1; order2=8; avisit='Day -1'; occasion=''; output; 

    order1=2; avisitn=100; paramn=1; order2=1; avisit='Day 0'; occasion=''; output; 

    order1=2; avisitn=100; paramn=1; order2=5; avisit='Day 0'; occasion=''; output; 

    order1=2; avisitn=100; paramn=1; order2=6; avisit='Day 0'; occasion=''; output; 



    order1=2; avisitn=100; paramn=1; order2=7; avisit='Day 0'; occasion=''; output; 

    order1=2; avisitn=100; paramn=1; order2=8; avisit='Day 0'; occasion=''; output; 

    order1=2; avisitn=101; paramn=1; order2=1; avisit='Day 1';occasion='Randomization Period'; output; 

run; 

 

data final; 

    merge  shell final1; 

    by order1 avisitn avisit paramn order2; 

 

    if order1=1 and order2 in (2 3 4) then do; stat='n (%)'; avisit='Day -2'; end; 

 

    if      paramn=1 then param='THSm2.2'; 

    else if paramn=11 then param='mCC'; 

 

    if not first.paramn then param=''; 

 

    if paramn=1 and avisitn=98 then do; 

        if      order2=2 then param='1'; 

        else if order2=3 then param='2'; 

        else if order2=4 then param='3'; 

        else if order2>4 then param=''; 

    end; 

 

    if order1=1 and order2=4 then do; 

        if col1='' then col1='0'; if col2='' then col2='0';   



        if col3='' then col3='0'; if col4='' then col4='0';   

    end; 

    else if order1=2 and order2=1 then do; 

        if col1='' then col1='0'; if col2='' then col2='0';   

        if col3='' then col3='0'; if col4='' then col4='0';   

    end; 

 

    if      order2=1 then stat='n'; 

    else if order2=5 then stat='Mean (SD)'; 

    else if order2=6 then stat='95% CI'; 

    else if order2=7 then stat='Median'; 

    else if order2=8 then stat='Min, Max'; 

 

    if avisitn in (99 100) and paramn=1 and order2=1 then do;  

        col1='0'; col2='0'; col3='0'; col4='0';  

    end; 

 

    if      _n_<19 then pageno=1; 

    else if _n_<39 then pageno=2; 

    else if _n_<59 then pageno=3; 

    else if _n_<79 then pageno=4; 

    else if _n_<99 then pageno=5; 

run; 

 

%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.)); 



 

%global totalpage; 

 

data _null_; 

  set final end=eof; 

  if eof then do; 

    call symput('totalpage', trim(left(put(pageno,8.)))); 

  end; 

run; 

 

%put totalpage=&totalpage; 

 

%macro reppart; 

 

  %do i = 1 %to &totalpage; 

 

proc report data=final(where=(pageno=&i.))  headskip headline spacing=4 nowd split='|' 

style=[outputwidth=100%] style(header column)=[protectspecialchars=off]; 

    column pageno order1 occasion avisitn avisit paramn param order2 stat col1-col4; 

 

    define pageno  /order order=internal noprint; 

    define order1  /order order=internal noprint; 

    define avisitn /order order=internal noprint; 

    define paramn  /order order=internal noprint; 

    define order2  /order order=internal noprint; 

    define occasion /display "Occasion"         style(column)=[cellwidth=20% asis=on] style(header)=[just=l]; 



    define avisit  /order "Visit|(Study Day)" style(column)=[cellwidth=10% asis=on] style(header)=[just=l]; 

    define param   /display "Product|Use"       style(column)=[cellwidth=10% asis=on] 

style(header)=[just=l]; 

    define stat    /display "Statistic"         style(column)=[cellwidth=10% asis=on] style(header)=[just=l]; 

    define col1    /display "THSm2.2|(N=&n1)"          flow style(column)=[cellwidth=10% just=c]; 

    define col2    /display "mCC|(N=&n2)"              flow style(column)=[cellwidth=10% just=c]; 

    define col3    /display "SA|(N=&n3)"               flow style(column)=[cellwidth=10% just=c]; 

    define col4    /display "Overall PP Set|(N=&n4)"   flow style(column)=[cellwidth=12% just=c]; 

 

    compute before paramn ; 

        line ""; 

    endcomp; 

 

    compute before _page_ /style=[fontweight=bold fontsize=3.75]; 

        line @1 "&title1 &title2"; 

        line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  "; 

    endcomp; 

    compute after _page_/style=[fontsize=1.75]; 

        line @1 "Note: mCC = Menthol conventional cigarettes; SA = Smoking abstinence; THSm2.2 = 

Tobacco Heating System 2.2 Menthol."; 

        line @1 "Note: Percentages are based on the number of subjects indicated in the column header 

(N)."; 

        line @1 "Note: Confinement period is defined as period from Day -2 to Day 6."; 

        line @1 "Note: n indicates the number of reporting subjects."; 

/*        line @1 "";*/ 

        line @1 "&APPENDIX."; 



        line @1 "Study ID:ZRHM-REXA-07-JP         Program: &fprgname..sas        Status: 

&repversion./&fdate.          Page: &i of &totalpage."; 

    endcomp; 

 

    compute after pageno ; 

/*        line "";*/ 

    endcomp; 

run; 

 

%end; 

 

%mend; 

%reppart; 

 

 

ods listing; 

ods rtf close; 

 

data odata.%sysfunc(scan(&prgname,1,'_'));; 

    set final(in=a); 

    if a then group="PPROT1FL"; 

run; 

 


