
**********************************************************************; 

* Project           : ZRHM-REXA-07-JP 

* 

* Program name      : t1502020302_ZRHM-REXA-07_V1.sas 

* 

* Author            : W. Yang  

* 

* Date created      : 05/28/2015 

* 

* Purpose           : Create Table 15.2.2.3.2 Summary of Average Daily Product Use in Ambulatory Period 

� PP Set 

* 

* Revision History  : 

* 

* Date        Author      Ref    Revision (Date in YYYYMMDD format)  

* 

**********************************************************************; 

 

%let prgname=T1502020302_ZRHM_REXA_07_JP_V1; 

options nomprint nosymbolgen nomlogic validvarname=upcase; 

options sasautos=("W:\pmp07\macros" sasautos) notes; 

%init(delivery=9); 

%titlecsv(prgname=&prgname.); 

%put &title1; 

%put &title2; 

%put &APPENDIX; 



%put &endpoint; 

%put &outname.; 

 

data adsl; 

    set adam.adsl(where=(index(disccat,'Period 1')=0)); 

    if      trt01an=4 then do; trt=1; output; end; 

    else if trt01an=5 then do; trt=2; output; end; 

    else if trt01an=3 then do; trt=3; output; end; 

    trt=4; output;  

run ; 

 

proc sort data=adam.adex out=adex; 

    by usubjid avisitn avisit;  

    where dtype='AVERAGE'; 

run; 

 

data adex; 

    set adex; 

    if parcat3='AVERAGE DAILY PRODUCT USE IN AMBULATORY' and pprot2fl='Y' and pprot3fl='Y' and 

pprot4fl='Y' then do; 

        apuper=5; 

        apuperc='Overall ambulatory period'; 

    end; 

run; 

 

%macro repeat(pop1fl=, sub=); 



 

proc freq data =adsl(where=(&pop1fl.='Y' %if &sub=5 %then and pprot2fl='Y' and pprot3fl='Y';)) noprint;  

  table trt/out=treatabt (rename =(count=total)drop=percent); 

run; 

 

Data _null_;        

  Set treatabt; 

  Call symput('n'||strip(put(trt, best.)),strip(put(total, best.)));     

Run;   

 

%put &n1 &n2 &n3 &n4; 

 

*** Prepare data for analysis ***; 

proc sql; 

    create table data0 as 

    select a.trt, b.* 

    from adsl as a cross join adex as b 

    where a.usubjid=b.usubjid and a.&pop1fl.='Y' and b.apuper=&sub. 

    ; 

quit; 

 

*** Macro for Continuous variables statistics ***; 

%macro mmeans(in_dsn=, class=, var=, order1=); 

proc means data=&in_dsn noprint nway alpha=0.05; 

    class trt &class; 



    var &var; 

    output out= stat1 n=_n mean=_mean std=_std median=_median min=_min max=_max lclm=_lclm 

uclm=_uclm; 

run; 

 

data stat2 (drop=_:); 

    length n meansd median minmax ci $50 fm1-fm3 $10; 

    merge stat1 treatabt; 

    by trt; 

    deci=0; 

 

    fm1=cats(12,".",strip(put(deci+1,3.))); 

    fm2=cats(12,".",strip(put(deci+2,3.))); 

    fm3=cats(12,".",strip(put(deci+0,3.))); 

 

    if _std>.  then _std=ceil(_std*100)/100; 

    if _uclm>. then _uclm=ceil(_uclm*10)/10; 

    if _lclm>. then _lclm=floor(_lclm*10)/10; 

    if _lclm=. and _std=0 and _mean>. then do; _lclm=_mean; _uclm=_mean; end; 

 

    if      _mean ne . and _std ne . then meansd=strip(putn(_mean, fm1))||" 

("||strip(putn(_std,fm2))||")"; 

    else if _mean ne . and _std eq . then meansd=strip(putn(_mean, fm1))||" (NA)"; 

    if _lclm    ne . then ci=strip(putn(_lclm,fm1))||", "||strip(putn(_uclm,fm1)); 

    if _min     ne . then minmax=strip(putn(_min,fm3))||", "||strip(putn(_max,fm3)); 

    if _n=total>. then n=strip(put(_n,best12.))||' (100)'; 



    else if _n>.  then n=strip(put(_n,best12.))||' ('||put(_n/total*100,5.1)||')'; 

    if total    ne . then tot=strip(put(total,best12.)); 

    if _median  ne . then median=strip(putn(_median, fm1)); 

run; 

 

proc sort; by &class. trt; run; 

 

proc transpose data=stat2 out=stat3 prefix=col; 

    by &class; 

    var tot n meansd ci median minmax; 

    id trt; 

run; 

 

data final_&order1. (drop=_name_); 

    length stat $100; 

    set stat3;   

    if _name_='TOT'    then do; stat="N";         order2 =1; end; 

    if _name_='N'      then do; stat="n (%)";     order2 =2; end; 

    if _name_='MEANSD' then do; stat="Mean (SD)"; order2 =3; end; 

    if _name_='CI'     then do; stat="95% CI";    order2 =4; end; 

    if _name_='MEDIAN' then do; stat="Median";    order2 =5; end; 

    if _name_='MINMAX' then do; stat="Min, Max";  order2 =6; end; 

    order1=&order1; 

run; 

%mend mmeans; 



 

%mmeans(in_dsn=data0, class=paramn param apuper apuperc, var=aval, order1=&sub); 

%mend; 

 

%repeat(pop1fl=pprot2fl, sub=2); 

%repeat(pop1fl=pprot3fl, sub=3); 

%repeat(pop1fl=pprot4fl, sub=4); 

%repeat(pop1fl=pprot4fl, sub=5); 

 

data final; 

    set final_:; 

    param=tranwrd(param,'Ave. Daily ',''); 

    param=tranwrd(param,'Average Daily ',''); 

    param=tranwrd(param,' by Period',''); 

    param=tranwrd(param,' in Ambulatory',''); 

    param=strip(param); 

    if param='THS 2.2' then param='THSm2.2 Sticks'; 

    if param='mCC'     then param='CC/mCC'; 

 

    if order2=1 then do; 

        if col1='' then col1='NA'; if col2='' then col2='NA';   

        if col3='' then col3='NA'; if col4='' then col4='NA';   

    end; 

run; 

 



proc sort data=final; by param paramn apuper apuperc; run; 

 

data final(rename=(ord=paramn)); 

    set final; 

    by param paramn apuper apuperc; 

    retain ord; 

    if first.param then ord=paramn; 

    drop paramn; 

run; 

 

proc sort data=final; by paramn param apuper apuperc; run; 

 

data final; 

    set final; 

    pageno=ceil(_n_/24);; 

run; 

 

data odata.%sysfunc(scan(&prgname,1,'_'));; 

    set final; 

run; 

 

%global totalpage; 

 

data _null_; 

  set final end=eof; 



  if eof then do; 

    call symput('totalpage', trim(left(put(pageno,8.)))); 

  end; 

run; 

 

%put totalpage=&totalpage; 

 

%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.)); 

options nomprint nosymbolgen; 

 

%macro reppart; 

 

  %do i = 1 %to &totalpage; 

 

proc report data=final(where=(pageno=&i.))  headskip headline spacing=4 nowd split='|' 

style=[outputwidth=100%] style(header column)=[protectspecialchars=off]; 

    column pageno paramn param apuper apuperc order2 stat col1-col3; 

 

    define pageno  /order order=internal noprint; 

    define paramn  /order order=internal noprint; 

    define apuper  /order order=internal noprint; 

    define order2  /order order=internal noprint; 

    define param   /order "Product"                         style(column)=[cellwidth=15% asis=on] 

style(header)=[just=l]; 

    define apuperc  /order "Product Use|Time Periods"       style(column)=[cellwidth=20% asis=on] 

style(header)=[just=l]; 



    define stat    /display "Statistic"                     style(column)=[cellwidth=10% asis=on] 

style(header)=[just=l]; 

    define col1    /display "THSm2.2"          flow style(column)=[cellwidth=12% just=c]; 

    define col2    /display "mCC"              flow style(column)=[cellwidth=12% just=c]; 

    define col3    /display "SA"               flow style(column)=[cellwidth=12% just=c]; 

/*    define col4    /display "Overall PP Set"   flow style(column)=[cellwidth=12% just=c];*/ 

 

    compute before apuper ; 

/*        line "";*/ 

    endcomp; 

 

    compute before _page_ /style=[fontweight=bold fontsize=3.75]; 

        line @1 "&title1 &title2"; 

        line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  "; 

    endcomp; 

    compute after _page_/style=[fontsize=1.75]; 

        line @1 "Note: mCC = Menthol conventional cigarettes; SA = Smoking abstinence; THSm2.2 = 

Tobacco Heating System 2.2 Menthol; NA = Not applicable."; 

        line @1 "Note: Ambulatory periods defined as Period 2 ([Day 6 ambulatory � Day 30 Visit]), Period 

3 (]Day 30 Visit � Day 60 Visit]) and Period 4 (]Day 60 Visit � Day 90 Visit])."; 

        line @1 "Note: Percentages are based on the number of subjects indicated in each product use 

period (N)."; 

/*        line @1 "";*/ 

        line @1 "&APPENDIX."; 

        line @1 "Study ID:ZRHM-REXA-07-JP         Program: &fprgname..sas        Status: 

&repversion./&fdate.          Page: &i of &totalpage."; 

    endcomp; 



 

/*    compute after pageno ;*/ 

/*        line "";*/ 

/*    endcomp;*/ 

run; 

%end; 

 

%mend; 

%reppart; 

 

 

ods listing; 

ods rtf close; 

 


