
%let prgname=T1502030103_ZRHM_REXA_07_JP_V1; 

options mprint; 

ods escapechar='^'; 

 

options sasautos=("W:\pmp07\macros" sasautos) notes; 



%init(delivery=9); 

%titlecsv(prgname=&prgname.,version=3); 

 

options missing=""; 

 

data adbx1; 

  set adam.adbx; 

  where anl02fl='Y'; 

  if paramcd='CARBXHGB' and atpt='DAY 5 - 20:00 - 21:30' and COMPP1FL='Y' then output; 

  if paramcd='UMHBMCRE' and upcase(avisit)='DAY 5' and COMPP1FL='Y' then output; 

  if paramcd='U3HPMCRE' and upcase(avisit)='DAY 5' and COMPP1FL='Y' then output; 

  if paramcd='USPMACRE' and upcase(avisit)='DAY 5' and COMPP1FL='Y' then output; 

  if paramcd='UNNALCRE' and upcase(avisit)='DAY 90' and COMPP4FL='Y' then output; 

run;   

 

data adbx2; 

  set adbx1; 

  logaval=log(AVAL); 

  logbase=log(BASE);  

 

  if paramcd='CARBXHGB' then ord=4; 

  if paramcd='UMHBMCRE' then ord=1; 

  if paramcd='U3HPMCRE' then ord=2; 

  if paramcd='USPMACRE' then ord=3; 

  if paramcd='UNNALCRE' then ord=5; 



run; 

 

%macro doit; 

 

%trtrtfg(pgmname=&loutname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.)); 

 

%do i=1 %to 5; 

 

%if &i=5 %then %let getppfl=4; %else %let getppfl=1;;; 

 

proc sort data=adbx2 out=adbx2&i nodupkey; 

by ord; 

where ord=&i; 

run; 

 

data adbx2&i; 

  set adbx2&i; 

  call symputx("param&i",param); 

  call symputx("day&i",avisit); 

  keep paramcd param avisit ord; 

run; 

 

title1 bold j=l "&ltitle1 &title2"; 

title2 "The where clause used on the dataset adam.adbx: compp&getppfl.fl='Y' and anl02fl='Y'"; 

title3 "param: &&param&i, avisit: &&day&i" ; 



 

proc glm data=adbx2;  

where ord=&i; 

class trtp sex UCPDGR1; 

model logaval = logbase sex UCPDGR1 trtp; 

lsmeans trtp / pdiff =control('mCC') alpha=0.05 cl adjust=t; 

lsmeans trtp / pdiff =control('SA') alpha=0.05 cl adjust=t; 

ods output LSMeans=pval&i (keep=ProbtDiff trtp where=(TRTP="THSm2.2")); *p-value; 

ods output LSMeanCL=LSMeanCL&i (keep=trtp lowercl uppercl lsmean); *lsmean, C.I. for each arm; 

ods output LSMeanDiffCL=LSMeanDiffCL&i (keep=trtp _trtp lowercl uppercl difference 

where=(TRTP="THSm2.2")); *lsmean and C.I. for ratios; 

ods output FitStatistics=ROOTMSE&i (keep=rootmse); *RootMSE; 

run;quit; 

 

proc freq data=adbx2 noprint;  

where ord=&i and logaval>.; 

tables trtp/out=countn&i; 

run; 

 

data countn&i.1(keep=ord row col1) countn&i.2(keep=ord row col2); 

 set countn&i; 

  length col1 col2 $20; ord=&i; 

 

  if trtp='THSm2.2' then do; 

   row=2; col1=put(count,3.); output countn&i.1; 

  end; 



  if trtp='mCC' then do; 

   row=2; col2=put(count,3.); output countn&i.2; 

  end; 

run; 

 

data LSMeanCL&i.1(keep=ord row col1) LSMeanCL&i.2(keep=ord row col2); 

 set LSMeanCL&i(obs=3); 

  lsmean  = exp(lsmean); /* Ratio of geometric mean */ 

  LowerCL   = exp(LowerCL);   /* 95% CI lower bound */ 

  UpperCL   = exp(UpperCL);   /* 95% CI upper bound */  

  length col1 col2 $20; ord=&i; 

 

  if trtp='THSm2.2' then do; 

   row=3; col1=put(round(lsmean,0.01),7.2); output LSMeanCL&i.1; 

   row=4; col1=put(round(floor(lowercl*100)/100,0.01),7.2)||', 

'||strip(put(round(ceil(uppercl*100)/100,0.01),7.2)); output LSMeanCL&i.1; 

  end; 

  else if trtp='mCC' then do; 

   row=3; col2=put(round(lsmean,0.01),7.2); output LSMeanCL&i.2; 

   row=4; col2=put(round(floor(lowercl*100)/100,0.01),7.2)||', 

'||strip(put(round(ceil(uppercl*100)/100,0.01),7.2)); output LSMeanCL&i.2; 

  end; 

 

run; 

 

data LSMeanDiffCL&i; 



  set LSMeanDiffCL&i; 

  difference  = 100*exp(difference); /* Ratio of geometric mean */ 

  LowerCL   = 100*exp(LowerCL);   /* 95% CI lower bound */ 

  UpperCL   = 100*exp(UpperCL);   /* 95% CI upper bound */  

 

  length col3 $20; ord=&i; 

 

  if _trtp='mCC' then do; 

   row=3; col3=put(round(difference,0.01),7.2); output; 

   row=4; col3=put(round(floor(lowercl*100)/100,0.01),7.2)||', 

'||strip(put(round(ceil(uppercl*100)/100,0.01),7.2)); output; 

  end; 

 

  keep ord row col3 difference; 

 

run; 

 

data ROOTMSE&i; 

  set ROOTMSE&i; 

  length col3cv $7; ord=&i; 

  CVperc=100*sqrt(exp(RootMSE**2)-1); 

  row=3; col3cv=put(round(ceil(CVperc*100)/100,0.01),7.2); 

  keep ord row col3cv; 

run; 

 

data pval&i; 



  set pval&i(obs=1); 

  set LSMeanDiffCL&i(where=(row=3) keep=row difference); 

  length col3 $20; ord=&i; 

  row=5;  

 if (difference < 100) then Probt1=ProbtDiff/2; 

  else Probt1=1-ProbtDiff/2; 

  col3=put(Probt1,pvalue6.3); 

  keep ord row col3; 

run; 

 

data struct&i; 

set adbx2&i; 

length hd $50 stat $40; 

row=1; hd=param; output; 

row=2; hd=propcase(avisit); stat='n'; output; 

row=3; hd=''; stat='Geometric LS Mean (CV%)';output; 

row=4; hd=''; stat='95% CI';output; 

row=5; hd=''; stat='p-value (one-sided)';output; 

keep ord row hd stat; 

run; 

 

data res&i; 

  merge struct&i countn&i.1 countn&i.2 LSMeanCL&i.1 LSMeanCL&i.2 LSMeanDiffCL&i ROOTMSE&i 

pval&i; 

  by ord row; 

run; 



 

data res&i; 

  set res&i; 

  if row=3 then col3=trim(col3)||' ('||strip(col3cv)||')'; 

run; 

 

%end; 

 

ods rtf close; 

 

data final; 

  set res1-res5; 

  by ord row; 

  pagen=ceil(ord/2); 

  drop difference col3cv; 

run; 

 

data final; 

  set final end=eof; 

  by pagen; 

  if eof then call symputx("totalpage",pagen); 

run; 

 

data odata.&prgname.; 

set final; 



run; 

 

%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.)); 

%do i=1 %to &totalpage; 

 

title; footnote; 

/* 

title1 bold j=l "&title1 &title2"; 

title2 " "; 

 

 

footnote1 bold h=12pt 

"____________________________________________________________________________________

__________"; 

footnote2 j=l h=9pt 'Note: Adjusted geometric least squares (LS) means and confidence intervals (CIs) 

from an ANCOVA model conducted on log-transformed ' 

'values with log-transformed baseline value, study arm, sex and mCC consumption reported at screening 

as fixed effect factors. Geometrical CV% of ' 

'the ratio is estimated from the residual mean squares.'; 

footnote3 j=l h=9pt "Note: mCC = Menthol conventional cigarettes; THSm2.2 = Tobacco Heating System 

2.2 Menthol."; 

footnote4 j=l h=9pt "Note: p-value for one-sided test for comparison between products."; 

footnote5 j=l h=9pt "Note: For the primary objective, Total NNAL is evaluated at Day 90 while the other 

biomarkers are evaluated at Day 5. For the secondary objective, " 

"Total NNAL is evaluated at Day 5 while the other biomarkers are evaluated at Day 90."; 

footnote6 j=l h=9pt " "; 

footnote7  h=9pt j=l "&APPENDIX."; 



footnote8  h=9pt j=l "Study ID:ZRHM-REXA-07-JP         Program: &prgname..sas        Status: 

&repversion./&fdate.          Page: &i. of &totalpage"; 

*/ 

 

proc report data=final  headskip headline nowd split='~' style=[outputwidth=100%] style(header 

column)=[protectspecialchars=off]; 

where pagen=&i; 

  column pagen ord hd stat col1 col2 col3; 

  define pagen  /order order=internal noprint; 

  define ord  /order order=internal noprint; 

  define hd   /display "Variable" style(column)=[cellwidth=18% just=l] style(header)=[just=l]; 

  define stat   /display "Statistic" style(column)=[cellwidth=18% just=l] style(header)=[just=l]; 

  define col1   /display "THSm2.2" style(column)=[cellwidth=14% just=c]; 

  define col2   /display "mCC" style(column)=[cellwidth=14% just=c]; 

  define col3   /display "THSm2.2 : mCC~Ratio (%)" style(column)=[cellwidth=14% just=c]; 

 

 

COMPUTE before pagen; 

LINE @1 " "; 

ENDCOMP; 

 

COMPUTE after ord; 

LINE @1 ""; 

ENDCOMP; 

 

compute before _page_ /style=[fontweight=bold fontsize=3.75]; 



line @1 "&title1 &title2"; 

line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  "; 

endcomp; 

 

compute after _page_/style=[fontsize=1.75]; 

line @1'Note: Adjusted geometric least squares (LS) means and confidence intervals (CIs) from an 

ANCOVA model conducted on log-transformed ' 

'values with log-transformed baseline value, study arm, sex and mCC consumption reported at screening 

as fixed effect factors. Geometrical CV% of ' 

'the ratio is estimated from the residual mean squares.'; 

line @1 "Note: mCC = Menthol conventional cigarettes; THSm2.2 = Tobacco Heating System 2.2 

Menthol."; 

line @1 "Note: p-value is for the one-sided test for comparison between THSm2.2 and mCC. P-value at 

Day 90 is evaluated only if P-value at Day 5 " 

"is significant, in all biomarkers except for Total NNAL."; 

line @1 "Note: For the primary objective, Total NNAL is evaluated at Day 90 while the other biomarkers 

are evaluated at Day 5. For the secondary objective, " 

"Total NNAL is evaluated at Day 5 while the other biomarkers are evaluated at Day 90."; 

line @1  " "; 

line @1 "&APPENDIX."; 

line @1 "Study ID:ZRHM-REXA-07-JP         Program: &fprgname..sas        Status: &repversion./&fdate.          

Page: &i. of &totalpage"; 

 

endcomp; 

 

 

run; 

%end; 



ods listing; 

ods rtf close; 

%mend; 

 

%doit; 

 


