
/******************************************************************************                                         
*  Project          : ZRHM-REXA-07-JP                                                                                   
*  Program name     : T15020501_ZRHM_REXA_07_JP_V1.sas                                                                  
*  Author           : C. Liu                                                                                            
*  Date created     : 06/11/2015                                                                                        
*  Purpose          : Summary of Compliance as Measured by Exhaled CO (ppm) During Confinement Period – FAS in the SA Ar
m                                                                                                                       
*  Revision History                                                                                                     
*  Date        Author      Ref    Revision                                                                              
*******************************************************************************/                                        
                                                                                                                        
%let prgname=T15020501_ZRHM_REXA_07_JP_V1;                                                                              
options mprint;                                                                                                         
                                                                                                                        
options sasautos=("W:\pmp07\macros" sasautos) notes;                                                                    
%init(delivery=9);                                                                                                      
                                                                                                                        
%titlecsv(prgname=&prgname.);                                                                                           
                                                                                                                        
%put &title1;                                                                                                           
%put &title2;                                                                                                           
%put &APPENDIX;                                                                                                         
%put &endpoint;                                                                                                         
%put &outname.;                                                                                                         
                                                                                                                        
options missing="";                                                                                                     
                                                                                                                        
title;                                                                                                                  
footnote;                                                                                                               
                                                                                                                        
proc format;                                                                                                            
  value $ordf                                                                                                           
   '<=10' = '01'                                                                                                        
   '>10' = '02'                                                                                                         
;                                                                                                                       
run;                                                                                                                    
                                                                                                                        
proc sql noprint;                                                                                                       
  select n(usubjid) into :n                                                                                             
  from adam.adsl                                                                                                        
  where fasfl='Y' and trt01an=3;                                                                                        
quit;                                                                                                                   
%put &n;                                                                                                                
data adbx;                                                                                                              
  set adam.adbx(where=(fasfl='Y' and trtan=3 and paramcd='CO' and 102<=avisitn<=105 and atptn ne .));                   
  atptn=100*(atptn-int(atptn));                                                                                         
  avisit=propcase(avisit);                                                                                              
run;                                                                                                                    
                                                                                                                        
proc sort;                                                                                                              
  by avisitn avisit atptn;                                                                                              
run;                                                                                                                    
                                                                                                                        
proc freq data=adbx noprint;                                                                                            
  table avisitn*avisit*atptn/out=freqn outpct;                                                                          
run;                                                                                                                    
                                                                                                                        
data freqn;                                                                                                             
  set freqn;                                                                                                            
                                                                                                                        
  length result $20;                                                                                                    
                                                                                                                        
  result=put(count,8.0);                                                                                                
run;                                                                                                                    
                                                                                                                        
proc transpose data=freqn out=t_freqn;                                                                                  
  by avisitn avisit;                                                                                                    
  id atptn;                                                                                                             
  var result;                                                                                                           
run;                                                                                                                    
                                                                                                                        
proc freq data=adbx noprint;                                                                                            
  table avisitn*avisit*atptn*avalcat1/out=freq outpct;                                                                  
run;                                                                                                                    
                                                                                                                        
data freq1;                                                                                                             
  set freq;                                                                                                             
                                                                                                                        
  length result $20;                                                                                                    
                                                                                                                        
  result=strip(put(count,best.))||' ('||strip(put(pct_row,4.1))||')';                                                   
run;                                                                                                                    
                                                                                                                        
proc transpose data=freq1 out=freq2;                                                                                    
  by avisitn avisit avalcat1;                                                                                           
  id atptn;                                                                                                             
  var result;                                                                                                           



run;                                                                                                                    
                                                                                                                        
data all;                                                                                                               
  set t_freqn(in=a) freq2;                                                                                              
  by avisitn avisit;                                                                                                    
  length ord $8;                                                                                                        
  if a then do;                                                                                                         
    ord='00'; avalcat1='n';                                                                                             
  end;                                                                                                                  
  else do;                                                                                                              
    ord=put(avalcat1,$ordf.);                                                                                           
    avalcat1='n (%) '||strip(avalcat1)||' ppm';                                                                         
  end;                                                                                                                  
run;                                                                                                                    
                                                                                                                        
proc sort;                                                                                                              
  by avisitn avisit ord avalcat1;                                                                                       
run;                                                                                                                    
                                                                                                                        
data shell;                                                                                                             
  length avisit avalcat1 $40 ord $8;                                                                                    
                                                                                                                        
  avisitn=102; avisit='Day 2'; avalcat1='n'; ord='00'; output;                                                          
  avalcat1='n (%) <=10 ppm'; ord='01'; output;                                                                          
  avalcat1='n (%) >10 ppm'; ord='02'; output;                                                                           
  avisitn=103; avisit='Day 3'; avalcat1='n'; ord='00'; output;                                                          
  avalcat1='n (%) <=10 ppm'; ord='01'; output;                                                                          
  avalcat1='n (%) >10 ppm'; ord='02'; output;                                                                           
  avisitn=104; avisit='Day 4'; avalcat1='n'; ord='00'; output;                                                          
  avalcat1='n (%) <=10 ppm'; ord='01'; output;                                                                          
  avalcat1='n (%) >10 ppm'; ord='02'; output;                                                                           
  avisitn=105; avisit='Day 5'; avalcat1='n'; ord='00'; output;                                                          
  avalcat1='n (%) <=10 ppm'; ord='01'; output;                                                                          
  avalcat1='n (%) >10 ppm'; ord='02'; output;                                                                           
run;                                                                                                                    
                                                                                                                        
proc sort;                                                                                                              
  by avisitn avisit ord avalcat1;                                                                                       
run;                                                                                                                    
                                                                                                                        
data final;                                                                                                             
  merge all shell;                                                                                                      
  by avisitn avisit ord avalcat1;                                                                                       
  array col _15 _25 _35 _45;                                                                                            
                                                                                                                        
  do i=1 to dim(col);                                                                                                   
    if col(i)='' then col(i)='0';                                                                                       
  end;                                                                                                                  
run;                                                                                                                    
                                                                                                                        
/*for QC purpose*/                                                                                                      
data odata.t15020501;                                                                                                   
     set all;                                                                                                           
run;                                                                                                                    
                                                                                                                        
data final;                                                                                                             
  set final;                                                                                                            
  by avisitn avisit ord avalcat1;                                                                                       
  retain _page 1;                                                                                                       
  if first.AVISIT then _c+1;                                                                                            
  if _c=5 then do;                                                                                                      
    _page+1;                                                                                                            
    _c=1;                                                                                                               
  end;                                                                                                                  
                                                                                                                        
  avalcat1=tranwrd(avalcat1,'<=',"^{unicode 2264}");                                                                    
  drop _c;                                                                                                              
run;                                                                                                                    
                                                                                                                        
proc sort data=final;                                                                                                   
  by _page;                                                                                                             
run;                                                                                                                    
                                                                                                                        
%global totalpage;                                                                                                      
                                                                                                                        
data _null_;                                                                                                            
  set final end=eof;                                                                                                    
                                                                                                                        
  if eof then do;                                                                                                       
    call symput('totalpage', trim(left(put(_page,8.))));                                                                
  end;                                                                                                                  
run;                                                                                                                    
                                                                                                                        
%put totalpage=&totalpage;                                                                                              
                                                                                                                        
%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.));                                      
                                                                                                                        



%macro reppart;                                                                                                         
                                                                                                                        
  %do i = 1 %to &totalpage;                                                                                             
                                                                                                                        
proc report data=final headskip headline spacing=4 nowd split='+' style=[outputwidth=100%]                              
            style(header column)=[protectspecialchars=off];                                                             
     columns (_page AVISITN ('^S={just=l}Timepoint' AVISIT) ord ('^S={just=l}Statistic' avalcat1)                       
             ("^R'\brdrb\brdrs 'SA+(N=%cmpres(&n))" ('^S={just=l}8:00 AM-09:30 AM' _15) ('^S={just=l}12:00 PM-01:30 PM' 
_25)                                                                                                                    
             ('^S={just=l}04:00 PM-05:30 PM' _35) ('^S={just=l}08:00 PM-09:30 PM' _45)))                                
;                                                                                                                       
     where _page =&i.;                                                                                                  
                                                                                                                        
     define _page/group order=internal noprint;                                                                         
     define AVISITN/group order=internal noprint;                                                                       
     define AVISIT/group ' ' order=internal style(column)=[cellwidth=8% just=l];                                        
     define ord/group order=internal noprint;                                                                           
     define avalcat1/group ' ' order=internal style(column)=[cellwidth=8% just=l];                                      
     define _15/display ' ' style(column)=[just=c cellwidth=10%];                                                       
     define _25/display ' ' style(column)=[just=c cellwidth=10%];                                                       
     define _35/display ' ' style(column)=[just=c cellwidth=10%];                                                       
     define _45/display ' ' style(column)=[just=c cellwidth=10%];                                                       
                                                                                                                        
     compute after AVISITN;                                                                                             
       line ' ';                                                                                                        
     endcomp;                                                                                                           
     break after _page/page;                                                                                            
                                                                                                                        
compute before _page_ /style=[fontweight=bold fontsize=3.75];                                                           
line @1 "&title1 &title2";                                                                                              
line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  ";                                                                    
endcomp;                                                                                                                
                                                                                                                        
compute before _page;                                                                                                   
line @1 "";                                                                                                             
endcomp;                                                                                                                
                                                                                                                        
compute after _page_/style=[fontsize=1.75];                                                                             
line @1 "Note: SA = Conventional cigarettes; THSm2.2 = Tobacco Heating System 2.2 Menthol.";                            
line @1 "Note: Percentages are based on the number of subjects indicated in the column header (N).";                    
line @1 " ";                                                                                                            
line @1 "&APPENDIX.";                                                                                                   
line @1 "Study ID:ZRHM-REXA-07-JP              Program: &fprgname..sas             Status: &repversion./&fdate.         
      Page: &i. of &totalpage";                                                                                         
endcomp;                                                                                                                
                                                                                                                        
run;                                                                                                                    
%end;                                                                                                                   
                                                                                                                        
%mend;                                                                                                                  
%reppart;                                                                                                               
                                                                                                                        
ods listing;                                                                                                            
ods rtf close;                                                                                                          

