
/******************************************************************************                                         
*  Project          : ZRHM-REXA-07-JP                                                                                   
*  Program name     : t15020620_ZRHM_REXA_07_JP_v1.sas                                                                  
*  Author           : H. Shu                                                                                            
*  Date created     : 05/21/2015                                                                                        
*  Purpose          : Summarize supine vital signs data for the study.                                                  
*  Revision History                                                                                                     
*  Date        Author      Ref    Revision                                                                              
*  05/25/2015  HS                 Adjusted to naming convention                                                         
*  05/30/2015  HS                 Adjusted to revised naming convention and title/footnote formatting                   
*  06/11/2015  HS                 Adjusted based on review comments including new title/footnote formatting             
*  07/13/2015  HS                 Corrected QC data set                                                                 
*******************************************************************************/                                        
%let prgname=T15020620_ZRHM_REXA_07_JP_V1;                                                                              
options sasautos=("W:\pmp07\macros" sasautos);                                                                          
%init(delivery=9);                                                                                                      
%titlecsv(prgname=&prgname,version=3);                                                                                  
%trtrtfg(pgmname=&outname., pgmid=1, new=0, style=, bookmark=%lowcase(&outname.));                                      
                                                                                                                        
/*population - post-randomization safety*/                                                                              
data _adsl;                                                                                                             
     set adam.adsl;                                                                                                     
     where SAFAFL='Y';                                                                                                  
     if TRT01AN=3 then TRT01AN=6;                                                                                       
run;                                                                                                                    
                                                                                                                        
proc sort data=_adsl out=_p;                                                                                            
     by TRT01AN;                                                                                                        
quit;                                                                                                                   
                                                                                                                        
data _trxf(keep=fmtname start label);                                                                                   
     set _p end=lr;                                                                                                     
     by TRT01AN;                                                                                                        
     if first.TRT01AN then _c=.;                                                                                        
     _c+1;                                                                                                              
     _t+1;                                                                                                              
     fmtname='_trxf';                                                                                                   
     if last.TRT01AN then do;                                                                                           
        call symput('_'||left(put(TRT01AN,8.)),trim(left(put(_c,8.))));                                                 
        start=TRT01AN;                                                                                                  
        label='\brdrb\brdrs '||trim(left(TRT01A))||'+(N='||trim(left(put(_c,8.)))||')';                                 
        output;                                                                                                         
     end;                                                                                                               
     if lr then do;                                                                                                     
        call symput('_100',trim(left(put(_t,8.))));                                                                     
        start=100;                                                                                                      
        label='\brdrb\brdrs Overall Safety+(N='||trim(left(put(_t,8.)))||')';                                           
        output;                                                                                                         
     end;                                                                                                               
run;                                                                                                                    
                                                                                                                        
proc format cntlin=_trxf;                                                                                               
quit;                                                                                                                   
                                                                                                                        
/*VS*/                                                                                                                  
data _vs0;                                                                                                              
     set adam.advs;                                                                                                     
     where ((SAFAFL='Y' & AVISITN>100) | (SAFBFL='Y' & TRTAN in (3 4 5) & ABLFL='Y')) & VSPOS='SUPINE' & AVAL>=0 & ANL01
FL='Y' & DTYPE='';                                                                                                      
     if TRTAN=3 then TRTAN=6;                                                                                           
     _h=1;                                                                                                              
     _res=AVAL;                                                                                                         
     output;                                                                                                            
     _h=2;                                                                                                              
     _res=CHG;                                                                                                          
     output;                                                                                                            
     TRTAN=100;                                                                                                         
     _h=1;                                                                                                              
     _res=AVAL;                                                                                                         
     output;                                                                                                            
     _h=2;                                                                                                              
     _res=CHG;                                                                                                          
     output;                                                                                                            
run;                                                                                                                    
                                                                                                                        
proc sort data=_vs0;                                                                                                    
     by PARAMN PARAM AVALU AVISITN AVISIT TRTAN _h;                                                                     
quit;                                                                                                                   
                                                                                                                        
proc univariate data=_vs0 noprint;                                                                                      
     by PARAMN PARAM AVALU AVISITN AVISIT TRTAN _h;                                                                     
     var _res;                                                                                                          
     output out=_vs1 n=n mean=Mean std=SD median=Median min=Min max=Max;                                                
quit;                                                                                                                   
                                                                                                                        
proc transpose data=_vs1 out=_vs2;                                                                                      
     by PARAMN PARAM AVALU AVISITN AVISIT TRTAN _h;                                                                     



     var n Mean SD Median Min Max;                                                                                      
quit;                                                                                                                   
                                                                                                                        
/*for QC purpose*/                                                                                                      
data odata.t15020620;                                                                                                   
     set _vs2;                                                                                                          
run;                                                                                                                    
                                                                                                                        
data _vs3;                                                                                                              
     set _vs2;                                                                                                          
     by PARAMN PARAM AVALU AVISITN AVISIT TRTAN _h;                                                                     
     if first.AVISIT then _c+1;                                                                                         
     if _c=3 then do;                                                                                                   
        _page+1;                                                                                                        
        _c=1;                                                                                                           
     end;                                                                                                               
     drop _c;                                                                                                           
run;                                                                                                                    
                                                                                                                        
data _rpt(keep=_page _tp PARAMN PARAM PARAM_ AVISITN AVISIT _ord _name_ TRTAN _h start)                                 
     _rfmt(keep=start label fmtname);                                                                                   
     set _vs3;                                                                                                          
     by _page PARAMN PARAM AVALU AVISITN AVISIT TRTAN _h;                                                               
     format label $40.;                                                                                                 
     PARAM_='';                                                                                                         
     _tp='_tp';                                                                                                         
     start=_n_;                                                                                                         
     if AVISIT='Day 0' then AVISIT='Baseline';                                                                          
        else if AVISIT='Day 6/Discharge Confinement' then AVISIT='Day 6';                                               
             else if AVISIT='Day 91/Discharge Ambulatory' then AVISIT='Day 91';                                         
     if AVALU in ('BEATS/MIN' 'BREATHS/MIN') then AVALU=lowcase(AVALU);                                                 
     PARAM=trim(left(PARAM))||' ('||trim(left(AVALU))||')';                                                             
     select(_name_);                                                                                                    
       when ('N') do;                                                                                                   
                    _ord=1;                                                                                             
                    _name_='n';                                                                                         
                    if _h^=2 | AVISITN^=100 then label=trim(left(put(col1,8.)));                                        
                       else label='';                                                                                   
                  end;                                                                                                  
       when ('MEAN') do;                                                                                                
                       _ord=2;                                                                                          
                       _name_='Mean';                                                                                   
                       if _h^=2 | AVISITN^=100 then label=trim(left(put(col1,8.1)));                                    
                          else label='';                                                                                
                     end;                                                                                               
       when ('SD') do;                                                                                                  
                     _ord=3;                                                                                            
                     _name_='(SD)';                                                                                     
                     if _h^=2 | AVISITN^=100 then label='('||trim(left(put(ceil(100*col1)/100,best.)))||')';            
                        else label='';                                                                                  
                   end;                                                                                                 
       when ('MEDIAN') do;                                                                                              
                         _ord=4;                                                                                        
                         _name_='Median';                                                                               
                         if _h^=2 | AVISITN^=100 then label=trim(left(put(col1,8.1)));                                  
                            else label='';                                                                              
                       end;                                                                                             
       when ('MIN') do;                                                                                                 
                      _ord=5;                                                                                           
                      _name_='Min,';                                                                                    
                      if _h^=2 | AVISITN^=100 then label=trim(left(put(col1,8.1)))||',';                                
                         else label='';                                                                                 
                    end;                                                                                                
       when ('MAX') do;                                                                                                 
                      _ord=6;                                                                                           
                      _name_='Max';                                                                                     
                      if _h^=2 | AVISITN^=100 then label=trim(left(put(col1,8.1)));                                     
                         else label='';                                                                                 
                    end;                                                                                                
       otherwise;                                                                                                       
     end;                                                                                                               
     fmtname='_vsrpt';                                                                                                  
run;                                                                                                                    
                                                                                                                        
proc format cntlin=_rfmt;                                                                                               
quit;                                                                                                                   
                                                                                                                        
proc format;                                                                                                            
     value _sh 1='Raw+value'                                                                                            
               2='+Change';                                                                                             
quit;                                                                                                                   
                                                                                                                        
%macro bprpt;                                                                                                           
proc sql noprint;                                                                                                       
     select max(_page) into: _tp from _rpt;                                                                             
quit;                                                                                                                   
                                                                                                                        



%do _i=0 %to &_tp;                                                                                                      
proc report data=_rpt headskip headline spacing=4 nowd split='+' style=[outputwidth=100%]                               
            style(header column)=[protectspecialchars=off] missing;                                                     
     where _page=&_i;                                                                                                   
     columns _page PARAMN PARAM ('^S={just=l}Parameter+(units)' PARAM_) AVISITN ('^S={just=l}Study+Day' AVISIT) _ord ('^
S={just=l}Statistic' _name_) (TRTAN,(_h,start));                                                                        
     define _page/group order=internal noprint;                                                                         
     define PARAMN/group order=internal noprint;                                                                        
     define PARAM/group order=internal noprint;                                                                         
     define PARAM_/group ' ' order=internal style(column)=[cellwidth=20% just=l];                                       
     define AVISITN/group order=internal noprint;                                                                       
     define AVISIT/group ' ' order=internal style(column)=[cellwidth=8% just=l];                                        
     define _ord/group order=internal noprint;                                                                          
     define _name_/group ' ' order=internal style(column)=[cellwidth=8% just=l];                                        
     define TRTAN/across ' ' order=internal f=_trxf. style(column)=[just=c];                                            
     define _h/across ' ' order=internal f=_sh. style(column)=[just=c];                                                 
     define start/analysis ' ' f=_vsrpt. style(column)=[just=c cellwidth=7%];                                           
                                                                                                                        
     compute before PARAM_/style=[just=l];                                                                              
       line PARAM $40.;                                                                                                 
       line ' ';                                                                                                        
     endcomp;                                                                                                           
                                                                                                                        
     compute after AVISITN;                                                                                             
       line ' ';                                                                                                        
     endcomp;                                                                                                           
                                                                                                                        
     compute before _page;                                                                                              
       line ' ';                                                                                                        
     endcomp;                                                                                                           
                                                                                                                        
     break after _page/page;                                                                                            
                                                                                                                        
     compute before _page_ /style=[fontweight=bold fontsize=3.75];                                                      
       line @1 "&title1 &title2";                                                                                       
       line @1  "^R/RTF'\brdrb\brdrs\brdrw30\brsp20\b '  ";                                                             
     endcomp;                                                                                                           
                                                                                                                        
     compute after _page_/style=[fontsize=1.75];                                                                        
       line @1 "Note: mCC = Menthol conventional cigarettes; SA = Smoking abstinence; THSm2.2 = Tobacco Heating System 2
.2 Menthol.";                                                                                                           
       line @1 "Note: Change is change from baseline, where baseline is defined as the last assessment prior to first " 
               "randomized product use in mCC / THS 2.2 Menthol arms or the last assessment prior to 10 AM on Day 1 in t
he SA arm.";                                                                                                            
       line @1  " ";                                                                                                    
       line @1 "&APPENDIX.";                                                                                            
       line @1 "Study ID:ZRHM-REXA-07-JP         Program: &fprgname..sas        Status: &repversion./&fdate.          Pa
ge: %eval(&_i+1) of %eval(&_tp+1)";                                                                                     
     endcomp;                                                                                                           
run;                                                                                                                    
%end;                                                                                                                   
%mend;                                                                                                                  
%bprpt                                                                                                                  
                                                                                                                        
ods listing;                                                                                                            
ods rtf close;                                                                                                          

